The Coral Language Specification

Kateřina Markéta Lisová

August 14, 2014

Contents

Preface

Coral is a Ruby-like programming language which enhances advanced object-oriented programming with elements of functional programming. Everything is an object, in this sense it's a pure object-oriented language. Object blueprints are described by classes. Classes can be composed in multiple ways – classic inheritance and/or trait composition, along with prototype-oriented inheritance.

Coral is also a functional language in the sense that every function is also an object, and generally, everything is a value. Therefore, function definitions can be nested and higher-order functions are supported out-of-the-box. Coral also has a limited support for pattern matching, which can emulate the algebraic types used in other functional languages.

Coral has been developed since 2012 in a home environment out of pure enthusiasm for programming and out of a desire for a truly versatile language. This document is a work in progress and will stay that way forever. It acts as a reference for the language definition and some core library classes.

Some of the languages that had major influence on the development of Coral, including syntax and behavior patterns, are Ruby, Ada, Scala, Java, C#, F#, Clojure and ATS. Coral tries to inherit their good parts and put them together in its own way.

The vast majority of Coral's syntax is inspired by *Ruby*. Coral uses keyword program parentheses in Ruby fashion. There is **`class`** … **`end`**, **`def`** … **`end`**, **`do`** … **`end`**, **`loop`** … **`end`**. Ruby itself is inspired by other languages, so this relation is transitive and Coral is inspired by those languages as well (for example, Ada).

Coral is inspired by *Ada* in the way that user identifiers are formatted: `Some_Constant_Name` and — unlike in Ada, but quite similar to it — `some_method_name`. Also, some control structures are inspired by Ada, such as loops, named loops, return expressions and record types. Pretty much like in Ada, Coral's control structures can be usually ended the same way: **`class`** … **`end class`**, etc.

Scala influenced the type system in Coral. Syntax for existential types comes almost directly from it. However, Coral is a rather dynamically typed language, so the type checks are made eventually in runtime (but some limited type checks can be made during compile time as well). Also, along with other C-like languages, Scala influenced Coral to allow to choose between Ada-style program parentheses (e.g. **`begin`** … **`end`**) and C-style parentheses (i.e. `{` … `}`) in many places throughout the syntax. Moreover, the structure of this mere specification is inspired by Scala's specification.

From *F#*, Coral borrows some functional syntax (like function composition) and F# also inspired the feature of Units of Measure (§1.5.4 & §5.9).

Clojure inspired Coral in the way functions can get their names. Coral realizes that turning function names into sentences does not always work, so it is possible to use

dashes, plus signs and slashes inside of function names. Therefore, `call/cc` is a legit function identifier. Indeed, binary operators are required to be properly surrounded by whitespace or other non-identifier characters.

ATS inspired Coral with dependent types (§3.3.11 & §5.8).

Chapter 1

Lexical Syntax

Coral programs are written using the Unicode character set; Unicode supplementary characters are supported as well. Coral programs are preferably encoded with the UTF-8 character encoding. While every Unicode character is supported, usage of Unicode escapes is encouraged, since fonts that IDEs might use may not support the full Unicode character set.

Grammar of lexical tokens is given in the following sections. These tokens are then used as terminal symbols of the semantical grammar.

To construct tokens, characters are distinguished according to the following classes (Unicode general category given in parentheses):

- Whitespace characters. \u0020 | \u0009 | \u000D | \u000A (space, tab character, carriage return, line feed)
- Letters, which include lower case letters (Ll), upper case letters (Lu), title-case letters (Lt), other letters (Lo), letter numerals (Nl), modifier letters (Lm) and the two characters \u0024 '$' and \u005F '_', which both count as upper case letters.
- Digits '0' | … | '9'.
- Parentheses. '(' | ')' | '[' | ']' | '{' | '}'
- Delimiter characters. ''' | '"' | '.' | ';' | ','
- Operator characters. These consist of all printable ASCII characters \u0020-\u007F that are in none of the sets above, mathematical symbols (Sm) and other symbols (So).

1.1 Identifiers

Syntax:

```
simple_id     ::= (lower | '_') [id_rest]
variable_id   ::= simple_id | '_' | '`' simple_id '`'
ivar_id       ::= '@' simple_id
cvar_id       ::= '@@' simple_id
op_id         ::= opchar {opchar}
function_id   ::= simple_id [id_rest_fun]
                | '`' simple_id [id_rest_fun] '`'
constant_id   ::= upper [id_rest_con]
                | '`' upper [id_rest_con] '`'
id_rest       ::= {letter | digit | '_'}
id_rest_con   ::= id_rest [id_rest_mid]
id_rest_fun   ::= id_rest [id_rest_mid] [['_'] op_id]
id_rest_mid   ::= id_rest {('/' | '+' | '-') id_rest}
importable_id ::= simple_id
                | function_id
                | constant_id
                | '`' op_id '`'
```

There are three kinds of identifiers.

First, *variable identifiers*, which are simply a lower-case letter followed by arbitrary sequence of letters (any-case), digits and underscores, or just one underscore (which has special meaning). Additionally, *instance variable identifiers* are just prepended with a "`@`" sign and *class instance variable identifiers* are just prefixed with "`@@`".

Second, *function identifiers*, which are the most complicated ones. They can start as a variable identifier, then optionally followed by one of "`/`", "`+`" and "`-`", and then optionally ended with "`?`", "`!`" or "`=`". Furthermore, function identifiers ending with "`=`" are never used at call site with this last character, but without it and as a target of an assignment expression (they are naming simple setters).

And third, *constant identifiers*, which are just like function identifiers, but starting with an upper-case letter, never just an underscore and never ending with "`?`", "`!`" or "`=`".

An identifier may also be formed by an identifier between back-quotes ("` ` `"), to resolve possible name clashes with Coral keywords. Instance variable names (`ivar_id`) and class instance variable names (`cvar_id`) never clash with a keyword name, since these are distinguished by the preceding "`@`" and "`@@`" respectively.

Coral programs are parsed greedily, so that a longest match rule applies. Letters from the syntax may be any Unicode letters, but English alphabet letters are recommended, along with English names.

1.2 Keywords

A set of identifiers is reserved for language features instead of for user identifiers. However, unlike in most other languages, keywords are not being recognized inside of paths, except for a few specific cases.

The following names are the reserved words.

```
abstract    alias       annotation  as          begin
bitfield    break       case        cast        catch
class       clone       cloned      constant    constructor
declare     def         delta       destructor  digits
do          else        elsif       end         ensure
enum        extends     final       for         for-some
function    get         goto        if          indexed-with
implements  implicit    in          include     interface
is          label       lazy        let         loop
match       memoize     message     method      module
native      next        nil         no          object
of          opaque      operator    out         override
pragma      prepend     private     property    protected
protocol    public      raise       range       record
redo        refine      refinement  rescue      retry
return      requires    reverse     sealed      self
set         singleton-type          skip        step
super       then        this        throw       trait
transparent type        unless      until       union
unit-of-measure         use         val         var
yes         weak        when        while       with
yield
```

Not every reserved word is a keyword in every context, this behavior will be further explained. For example, the bitfield reserved word is only recognized as a keyword inside an enumeration definition context, in a specific place. Every reserved word may be used as a function identifier, with a little work-around when used with an implicit receiver.

1.3 Newline Characters

Syntax:

```
semi ::= nl {nl} | ';'
```

Coral is a line-oriented language, in which statements are expressions and may be

terminated by newlines, as well as by semi-colon operator. A newline in a Coral source file is treated as the special separator token `nl` if the following criterion is satisfied:

1. The token immediately preceding the newline can terminate an expression.

Since Coral may be interpreted in a REPL[1] fashion, there are no other suitable criteria. Such a token that can terminate an expression is, for instance, not a binary operator or a message sending operator, which both require further tokens to create an expression. Keywords that expect any following tokens also can not terminate expressions. Coral interpreters and compilers do not look-ahead beyond newlines.

If the token immediately preceding the newline can not terminate an expression and is followed by more than one newline, Coral still sees that as only a one significant newline, to prevent any confusion.

Keywords that can terminate an expression are: **`break`**, **`end`**, **`opaque`**, **`native`**, **`next`**, **`nil`**, **`no`**, **`redo`**, **`retry`**, **`return`**, **`self`**, **`skip`**, **`super`**, **`this`**, **`transparent`**, **`yes`**, **`yield`**.

1.4 Operators

A set of identifiers is reserved for language features, some of which may be overridden by user space implementations. Operators have language-defined precedence rules that are supposed to usually comply to user expectations (principle of least surprise), and another desired precedence may be obtained by putting expressions with operators inside of parenthesis pairs.

The following character sequences are the operators recognized by Coral.

```
:=     +=     -=     *=     **=    /=     %=     ||=    &&=    ^^=
|=     &=     |=     [      ^=     ~=     <<     >>     <<<    >>>
<<=    >>=    <<<=   >>>=   ;      =      !=     ==     !==    ===
!===   =~     !~     <>     <      >      <=     >=     <=>    +
-      *      **     /      div    %      mod    ||     or     &&
and    !      not    ^^     xor    |      &      ^      ~      ..
...    ,      ->     <-     ~>     <~     =>     ::     :      <:
:>     <<|    |>>    <|     |>     (      )      [      ]      {
}      .      [<     >]     >:     .?     .!     rem
```

Some of these operators have multiple meanings, usually up to two. Some are binary, some are unary, none is ternary.

Binary (infix) operators have to be separated by whitespace or non-letter characters on both sides, unary operators on left side – the right side is what they are bound to.

[1] Read-Eval-Print Loop

Unary operators are: `+`, `−`, `&`, `not`, `!` and `~`. The first three of these are binary as well. The `;` operator is used to separate expressions (see Newline Characters). Parentheses are postcircumfix operators. Coral has no postfix operators.

Coral allows for custom user-defined operators, but those have the lowest precedence and need to be parenthesized in order to express any precedence. Such custom operators can't be made of letter characters.

1.5 Literals

There are literals for numbers (including integer, floating point and complex), characters, booleans, strings, symbols, regular expressions and collections (including tuples, lists, dictionaries and bags).

Syntax:

```
literal ::= integer_literal
          | floating_point_literal
          | complex_literal
          | character_literal
          | string_literal
          | symbol_literal
          | regular_expression_literal
          | Collection_Literal
          | 'nil'
```

1.5.1 Integer Literals

Syntax:

```
integer_literal     ::= ['+' | '−'] (decimal_numeral
                      | hexadecimal_numeral
                      | octal_numeral
                      | binary_numeral)
decimal_numeral     ::= '0' | non_zero_digit {['_'] digit}
hexadecimal_numeral ::= '0x' | hex_digit {['_'] hex_digit}
digit               ::= '0' | non_zero_digit
non_zero_digit      ::= '1' | ··· | '9'
hex_digit           ::= '1' | ··· | '9' | 'a' | ··· | 'f'
octal_numeral       ::= '0' oct_digit {'_' oct_digit}
oct_digit           ::= '0' | ··· | '7'
binary_numeral      ::= '0b' bin_digit {['_'] bin_digit}
bin_digit           ::= '0' | '1'
```

Integers are usually of type `Number`, which is a class cluster of all classes that can represent numbers. Unlike Java, Coral supports both signed and unsigned integers directly. Usually integer literals that are obviously unsigned integers are automatically represented internally by a class that stores the integer unsigned, like `Integer_64_Unsigned`. Math operations on numbers are handled internally in such a way that the user does't need to worry about the actual types of the numbers — when an integer overflow would occur, the result is stored in a larger container type.

Underscores used in integer literals have no special meaning, other than to improve readability of larger literals, i.e., to separate thousands.

Integral members of the `Number` class cluster include the following container types.

1. `Integer_8` (-2^7 to $2^7 - 1$), alias `Byte`
2. `Integer_8_Unsigned` (0 to 2^8), alias `Byte_Unsigned`
3. `Integer_16` (-2^{15} to $2^{15} - 1$), alias `Short`
4. `Integer_16_Unsigned` (0 to 2^{16}), alias `Short_Unsigned`
5. `Integer_32` (-2^{31} to $2^{31} - 1$)
6. `Integer_32_Unsigned` (0 to 2^{32})
7. `Integer_64` (-2^{63} to $2^{63} - 1$), alias `Long`
8. `Integer_64_Unsigned` (0 to 2^{64}), alias `Long_Unsigned`
9. `Integer_128` (-2^{127} to $2^{127} - 1$), alias `Double_Long`
10. `Integer_128_Unsigned` (0 to 2^{128}), alias `Double_Long_Unsigned`
11. `Decimal` ($-\infty$ to ∞)
12. `Decimal_Unsigned` (0 to ∞)

The special `Decimal` & `Decimal_Unsigned` container types are also for storing arbitrary precision floating point numbers. All the container types are constants defined in the `Number` class and can be imported into scope if needed.

Moreover, a helper type `Number.Unsigned` exists, which can be used for type casting in cases where an originally signed number needs to be treated as unsigned.

Weak conformance applies to the inner members of `Number` class.

For use with range types, `Number.Integer` and `Number.Integer_Unsigned` exist, to allow constraining of the range types to integral numbers.

1.5.2 Floating Point Literals

Syntax:

```
float_literal ::= ['+' | '-'] non_zero_digit
                  {['_'] digit} '.' digit {['_'] digit}
                  [exponent_part] [float_type]
                | ['+' | '-'] digit {['_'] digit} exponent_part
                  [float_type]
                | ['+' | '-'] digit {['_'] digit} [exponent_part]
                  float_type
                | ['+' | '-'] '0x' hex_digit
                  {['_'] hex_digit} '.' hex_digit
                  {['_'] hex_digit} [float_type]
                | ['+' | '-'] '0b' bin_digit
                  {['_'] bin_digit} '.' bin_digit
                  {['_'] bin_digit} [float_type]
exponent_part ::= 'e' ['+' | '-'] digit {['_'] digit}
float_type    ::= 'f' | 'd' | 'q' | 'df'
```

Floating point literals are of type `Number` as well as integral literals, and have fewer container types. Compiler infers the precision automatically, unless the `float_type` part is present. Floating point literals that have `float_type` of "`dp`" are decimal fixed point literals. Also, floating point literals that are impossible to represent in binary form accurately are implicitly fixed point literals. From user's perspective, this is only an implementation detail.

1. `Float_32` (IEEE 754 32-bit precision), alias `Float`.
2. `Float_64` (IEEE 754 64-bit precision), alias `Double`.
3. `Float_128` (IEEE 754 128-bit precision).
4. `Decimal` ($-\infty$ to ∞).
5. `Decimal_Unsigned` (0 to ∞).

Letters in the exponent type and float type literals have to be lower-case in Coral sources, but functions that parse floating point numbers do support them being upper-case for compatibility.

1.5.3 Imaginary Number Literals

Syntax:

```
imaginary_literal ::= real_number_literal 'i'
complex_literal ::= real_number_literal ('+' | '−') imaginary_literal
    | imaginary_literal ('+' | '−') real_number_literal
real_number_literal ::= integer_literal | float_literal
number_literal ::= real_number_literal
    | imaginary_literal
    | complex_literal
```

1.5.4 Units of Measure

Coral has an addition to number handling, called *units of measure* (§5.9). Number instances can be annotated with a unit of measure to ensure correctness of arithmetic operations.

Syntax:

```
annotated_number ::= number_literal '[<' uom_expr '>]'
uom_expr         ::= Unit_Conv {',' Unit_Conv}
```

1.5.5 Character Literals

Syntax:

```
character_literal ::= '%'' (character | unicode_escape) '''
```

1.5.6 Boolean Literals

Syntax:

```
boolean_literal ::= 'yes' | 'no'
```

Both literals are members of type `Boolean`. The **`no`** literal has also a special behavior when being compared to **`nil`**: **`no`** equals to **`nil`**, while not actually being **`nil`**. Identity equality is indeed different, and **`no`** does not match in pattern matching (§8) as **`nil`** and vice versa. The implication is that both **`nil`** and **`no`** are false conditions in **`if`**-expressions.

1.5.7 String Literals

Syntax:

```
string_literal             ::= simple_string_literal
                             | interpolable_string_literal
simple_string_literal      ::= ''' {string_element} '''
```

```
string_element              ::= printable_char | char_escape_seq
interpolable_string_literal ::= '"' {int_string_element} '"'
int_string_element          ::= string_element | interpolated_expr
interpolated_expr           ::= '#{' expr '}'
```

String literals are members of the type `String`. Single quotes in simple string literals have to be escaped (`\'`) and double quotes in interpolable string literals have to be escaped (`\"`). Interpolated expression can be preceded only by an even number of escape characters (backslashes, `\`), so that the `#` does't get escaped. This is a special *requirement* for any Coral compiler.

1.5.8 Symbol Literals

Syntax:

```
symbol_literal       ::= simple_symbol | quoted_symbol
simple_symbol        ::= ':' simple_id
quoted_symbol        ::= simple_quoted_symbol | interpolable_symbol
simple_quoted_symbol ::= ':'' {string_element} '''
interpolable_symbol  ::= ':"' {int_string_element} '"'
```

Symbol literals are members of the type `Symbol`. They differ from String Literals in the way runtime handles them: while there may be multiple instances of the same string, there is always up to one instance of the same symbol. Unlike in Ruby, they do get released from memory when no code references to them anymore, so their object id (sometimes) varies with time. Coral does not require their ids to be constant in time.

1.5.9 Type Parameters

Syntax:

```
tp_id ::= '$' constant_id
```

Type parameters are not members of any type, rather they stand-in for a real type, like a variable which only holds types.

1.5.10 Regular Expression Literals

Syntax:

```
regexp_literal     ::= '%/' regexp_content_int '/' [regexp_flags]
                     | '%r/' regexp_content_int '/' [regexp_flags]
                     | '%r#' regexp_content '#' [regexp_flags]
                     | '%r~' regexp_content_int '~' [regexp_flags]
regexp_content_int ::= regexp_element_int {regexp_element_int}
```

```
regexp_element_int ::= string_element | int_string_element
regexp_content     ::= string_element {string_element}
regexp_flags       ::= printable_char {printable_char}
```

Regular expression literals are members of the type `Regular_Expression` with alias of `Regexp`.

1.5.11 Collection Literals

Collection literals are paired syntax tokens and as such, they are a kind of parentheses in Coral sources.

Syntax:

```
Collection_Literal ::= Tuple_Literal
                     | List_literal
                     | Dictionary_Literal
                     | Bag_Literal
Tuple_Literal      ::= '(' Exprs ')'
List_Literal       ::= '%' Collection_Flags '[' Exprs ']'
Dictionary_Literal ::= '%' Collection_Flags '{' Dict_Exprs '}'
Bag_Literal        ::= '%' Collection_Flags '(' Exprs ')'
Dict_Exprs         ::= Dict_Expr {',' Dict_Expr}
Dict_Expr          ::= Expr '=>' Expr
                     | simple_id ':' Expr
Collection_Flags   ::= printable_char {printable_char}
```

Tuple literals are members of the `Tuple` type family. List literals are members of the `List` type, usually `Array_List` with alias of `Array`. Dictionary literals are members of the `Dictionary` type with alias of `Map`, usually `Hash_Dictionary` with alias of `Hash_Map`. Bag literals are members of the `Bag` type, usually `Hash_Bag` or `Hash_Set`. Collection flags may change the actual class of the literal, along with some other properties, described in the following text.

List literal collection flags:

1. Flag `i` = **i**mmutable, makes the list frozen.
2. Flag `l` = **l**inked, makes the list a member of `Linked_List`.
3. Flag `w` = **w**ords, the following expressions are treated as words, converted to strings for each word separated by whitespace.

Dictionary literals collection flags:

1. Flag `i` = **i**mmutable, makes the dictionary frozen.

2. Flag `l` = **l**inked, makes the dictionary a member of `Linked_Hash_Dictionary` (also has alias `Linked_Hash_Map`).

3. Flag `m` = **m**ulti-map, the dictionary items are then either the items themselves, if there is only one for a particular key, or a set of items, if there is more than one item for a particular key. The dictionary is then a member of `Multi_Hash_Dictionary` (alias `Multi_Hash_Map`) or `Linked_Multi_Hash_Dictionary` (alias `Linked_Multi_Hash_Map`).

Bag literal collection flags:

1. Flag `i` = **i**mmutable, makes the bag frozen.

2. Flag `s` = **s**et, the collection is a set instead of a bag (a specific bag, such that for each item, its tally is always 0 or 1, thus each item is in the collection up to once).

3. Flag `l` = **l**inked, makes the collection linked, so either a member of `Linked_Hash_Bag` in case of a regular bag, or `Linked_Hash_Set` in case of a set.

Linked collections have a predictable iteration order in case of bags and dictionaries, or are simply stored differently in case of lists.

1.6 Whitespace & Comments

Tokens may be separated by whitespace characters and/or comments. Comments come in two forms:

A single-line comment is a sequence of characters that starts with `//` and extends to the end of the line.

A multi-line comment is a sequence of characters between `/*` and `*/`. Multi-line comments may be nested.

Documentation comments are multi-line comments that start with `/*!`.

Chapter 2

Identifiers, Names & Scopes

Names in Coral identify various types, values, methods and constants, which are the *entities*. Names are introduced by local definitions and declarations, inheritance, use clauses or module clauses, which are the *bindings*.

Bindings of different kinds have a different precedence defined on them:

1. Definitions and declarations that are local have the highest precedence.
2. Explicit **`use`** clauses (imports) have the next highest precedence.[1]
3. Wildcard **`use`** clauses (imports) have the next highest precedence.
4. Inherited definitions and declarations have the next highest precedence.
5. Definitions and declarations made available by module clause have the next highest precedence.
6. Definitions and declarations that are not in the same compilation unit (a different script or a different module) have the next highest precedence.
7. Definitions and declarations that are not bound have the lowest precedence. This happens when the binding simply can't be found anywhere, and probably will result in a name error (if not resolved dynamically), while being inferred to be of type `Object`.

There is only one root name space, in which a single fully-qualified binding designates always up to one entity.

[1] Explicit imports have such high precedence in order to allow binding of different names than those that would be otherwise inherited.

Every binding has a *scope* in which the bound entity can be referenced using a simple name (unqualified). Scopes are nested, inner scopes inherit the same bindings, unless shadowed. A binding in an inner scope *shadows* bindings of lower precedence in the same scope (and nested scopes) as well as bindings of the same or lower precedence in outer scopes. Shadowing is a partial order, and bindings can become ambiguous – fully qualified names can be used to resolve binding conflicts. This restriction is checked in limited scope during compilation[2] and fully in runtime.

If at any point of the program execution a binding would change (e.g., by introducing a new type in a superclass that is closer in the inheritance tree to the actual class than the previous binding), and such a change would be incompatible with the previous binding, then a warning[3] will be issued by the runtime. Also, if a new binding would be ambiguous[4], then it is an error.

As shadowing is only a partial order, in a situation like

```
var x := 1
use p.x
x
```

neither binding of x shadows the other. Consequently, the reference to x on the third line above is ambiguous and the compiler will happily refuse to proceed.

A reference to an unqualified identifier x is bound by a unique binding, which

1. defines an entity with name x in the same scope as the identifier x, and
2. shadows all other bindings that define entities with name x in that name scope.

It is syntactically not an error if no such binding exists, thanks to the dynamic features of the language (unbound references are implicitly bound to the same scope and are resolved by dynamic method callbacks). The same applies to fully qualified bindings that don't resolve into any entity. However, it is an error if a binding is ambiguous or fails to get resolved dynamically.

If x is bound by explicit **use** import clause, then the simple name x is consided to be equivalent to the fully-qualified name to which x is mapped by the import clause. If x is bound by a definition or declaration, then x refers to the entity introduced by that binding, thus the type of x is the type of the referenced entity.

[2]This is due to the hybrid typing system in Coral, to make use of all the available information as soon as possible.

[3]TBD – shouldn't that be an error?

[4]Coral runtime actually checks for bindings until the binding-candidate would not be able to shadow the already found binding-candidates and caches the result.

Chapter 3

Types

Syntax:

```
Type                ::= Function_Type
                      | Compound_Type [Existential_Clauses]
Function_Type       ::= Function_Args {'->' Function_Args}
                          '->' Return_Type
Function_Args       ::= Compound_Type
                      | '(' [Function_Arg {',' Function_Arg}] ')'
Function_Arg        ::= [Param_Io] ['*'] [':' variable_id] Param_Type
                      | '&' Function_Type
Existential_Clauses ::= {'for-some' '{' Existential_Dcl
                          {semi Existential_Dcl} '}'}
Existential_Dcl     ::= 'type' Type_Dcl
                      | 'val' Val_Dcl
                      | 'var' Var_Dcl
Compound_Type       ::= Annot_Type {['prepend'] 'with' Annot_Type}
                          ['with' ['refinement'] Refine_Stats]
                      | ['refinement'] Refine_Stats
Annot_Type          ::= {Annotation} Simple_Type
Simple_Type         ::= Simple_Type [Type_Args] [Dep_Args] [UoM_Args]
                      | Simple_Type '#' constant_id
                      | Stable_Id
                      | Path '.' ('type' | 'singleton-type')
                      | '(' Types ')'
Types               ::= Type {',' Type}
```

When we say *type* in the context of Coral, we are talking about a blueprint of an entity, while the type itself is an entity. Every type in Coral is backed by a *class*, which is an instance of the type `Class`.

We distinguish a few different properties of types in Coral. There are first-order types

and type constructors, which take type parameters and yield new types. A subset of first-order types called *value types* represents set of first-class values. Value types are either *concrete* or *abstract*.

Concrete value types can be either a *class type* (e.g. referenced with a type designator, referencing a class or maybe a trait), or a *compound type* representing an intersection of types, possibly with a refinement that further constrains the types of its members. Both class types and compound types may be bound to a constant, but only class types referencing a concrete class can be blueprints of values – *objects*. Compound types can only constrain bindings to a subset of other types.

Non-value types capture properties of identifiers that are not values. For instance, a type constructor does not directly specify a type of values, but a type constructor, when applied to the correct type arguments, yields a first-order type, which may be a value type. Non-value types are expressed indirectly in Coral. In example, a method type is described by writing down a method signature, which is not a real type itself, but it creates a corresponding method type.

3.1 About Coral's Type System

There are two main streams of typing systems out there – statically typed and dynamically typed. Static typing in a language usually means that the language is compiled into an executable with a definite set of types and every operation is type checked. Dynamic typing means that these checks are deferred until needed, in runtime.

Let's talk about Java. Java uses static typing – but, in a very limited and unfriendly way, you may use class loaders and a lot of type casts to dynamically load a new class. And then possibly endure a lot of pain using it.

Let's talk about Ruby. Ruby uses dynamic typing – but, using types blindly can possibly lead to some confusion. Ruby is amazing though, because you can write programs with it really fast and enjoy the process at the same time. But when it comes to type safety, you need to be careful.

And now, move on to Coral. Coral uses hybrid typing. In its core, it uses dynamic typing all the way. But, it allows to opt-in for some limited static typing[1]. Unlike in Ruby, you can overload methods (not just override!). You can constrain variables, constants, properties, arguments and return types to particular types. But you don't have to. Types in Coral were heavily inspired by Scala's type system, but modified for this dynamic environment that Coral provides. Unlike in Ruby, you can have pure interfaces (called protocols[2]), or interfaces with default method implementations (similar to Java 8). Unlike in Java, you can have traits, union types and much more. Unlike in Java, you

[1]This feature is expected to be gradually improved and un-limited.

[2]Interfaces in Coral are used to extract the *public interface* of classes in modules, so that only a small amount of code may be distributed along with the module to allow binding to it.

may easily modify classes, even from other modules (*pimp my library!*). You may even easily add more classes if needed, and possibly shadow existing ones. In face of static typing in Coral, *no type* specified is saying that the value is of any type.

While Coral is so dynamic, it also needs to maintain stability and performance. Therefore, it "caches" its bindings and tracks versions of each type[3]. If a *cached binding* would change, it is ok – as long as the new binding would conform to the old one. Practically, the code that executes first initiates the binding – first to come, first to bind. Bindings are also cached, so that the Coral interpreter does not need to traverse types all the time – it only does so if the needed binding does not exist (initial state), or if the cached version does not match the actual version of the bound type. This mechanism is also used for caching methods, not only types. Moreover, this mechanism ensures that type projections (§3.3.2) are valid at any time of execution, even if their binding changes.

Types in Coral are represented by objects that are members of the `Class` type.

3.2 Paths

Syntax:

```
path_id          ::= constant_id
                   | variable_id
                   | function_id
                   | op_id
                   | path_id [Type_Args] [UoM_Args]
Path             ::= Stable_Id
                   | [path_id '.'] ('this' | 'self')
                   | 'self' '[' 'cloned' ']'
                   | Container_Path
Stable_Id        ::= path_id
                   | Path '.' path_id
                   | [path_id '.'] 'super' [Class_Qualifier]
                     '.' path_id
Class_Qualifier  ::= '[' path_id ']'
```

Paths are not types themselves, but they can be a part of named types and in that function form a role in Coral's type system.

A path is one of the following: [4]

[3]Versions are simply integers that are incremented with each significant change to the type and distributed among its subtypes.

[4]This section might need a review of what a path is, since we claim that the referenced entity is a member, yet the syntax only mentions `constant_id`.

- The empty path ϵ (which can not be written explicitly in user programs).
- **`this`**, which references the directly enclosing class.
- C**`.self`**, where C references a class or a trait. The path **`self`** is taken as a shorthand for C**`.self`**, where C is the name of the class directly enclosing the reference.
- **`self[cloned]`**, which references the directly enclosing class of a clone (a cloned instance, see §5.3.2).
- p**`.`**x, where p is a path and x is a member of p. Additionally, p allows modules to appear instead of references to classes or traits, but no module reference can follow a class or a trait reference: `{module_ref '.'} {(class_ref|trait_ref) '.'} ...`.
- C**`.super.`**x or C**`.super[`**M**`].`**x, where C references a class or a trait and x references a member of the superclass or designated parent class M of C. The prefix **`super`** is taken as a shorthand for C**`#super`**, where C is the name of the class directly enclosing the reference, and **`super[`**M**`]`** as a shorthand for C**`.super[`**M**`]`**, where C is yet again the name of the class directly enclosing the reference.

3.3 Value Types

Every value in Coral has a type which is of one of the following forms.

3.3.1 Value & Singleton Type

Syntax:

```
Simple_Type ::= Path '.' 'type'
Simple_Type ::= Path '.' 'singleton-type'
```

A singleton type is of the form p**`.singleton-type`** and a special type that denotes the set of values consisting of **`nil`** and the value denoted by p. A value type, on the other hand, is a special type that denotes the set of values consisting of **`nil`** and every value that conforms to the type of value denoted by p.[5]

3.3.2 Type Projection

Syntax:

```
Simple_Type ::= Simple_Type '#' constant_id
```

[5]This is useful when using a value as prototype of new values.

A type projection $T\#x$ references type member named x of type T. This is useful i.e. with nested classes that belong to the class instances, not the class object.

3.3.3 Type Designators

Syntax:

```
Simple_Type ::= Stable_Id
```

A type designator refers to a named value type. It can be simple or qualified. All such type designators are shorthands for type projections.

Specifically, the unqualified type name t where t is bound in some class, object or module C is taken as a shorthand for C`.self.type#`t. If t is not bound in a class, object or module, then t is taken as a shorthand for ϵ`.type#`t.

A qualified type designator has the form p`.`t, where p is a path (§3.2) and t is a type name. Such a type designator is equivalent to the type projection p`.type#`t.

3.3.4 Parameterized Types

Syntax:

```
Simple_Type ::= Simple_Type [Type_Args] [UoM_Args]
Type_Args   ::= '[' Types ']'
Types       ::= Type {',' Type}
UoM_Args    ::= '[<' uom_expr '>]'
```

A parameterized type $T[T_1, \ldots, T_n]$ consists of a type designator T and type parameters $T_1, \ldots, T_n$, where $n \geq 1$. T must refer to a type constructor which takes exactly n type parameters $a_1, \ldots, a_n$.

Say the type parameters have lower bounds $L_1, \ldots, L_n$ and upper bounds $U_1, \ldots, U_n$. The parameterized type is well-formed if each actual type parameter *conforms to its bounds*, so that $\sigma L_i <: T_i <: \sigma U_i$, where σ is the substitution $[a_1 := T_1, \ldots, a_n := T_n]$[6]. Also, U_i must never be a subtype of L_i, since no other type ever would be able to fulfil the bounds (U_i and L_i may be the exact same type though, but in that case the type parameter would be invariant and the whole point of having a parameterized type would be useless).

Example 3.3.1 Given the generic type definitions:

[6]The substitution works by replacing occurrences of a_i in the argument by T_i, so that, e.g. `$A <: Comparable[$A]` is substituted into C `<: Comparable[`C`]`.

```
class Tree_Map[$A <: Comparable[$A], $B] ... end
class List[$A] ... end
class I; implements Comparable[I]; ... end

class F[$M[_], $X] ... end
class S[$K <: String] ... end
class G[$M[$Z <: $I], $I] ... end
```

the following parameterized types are well-formed:

```
Tree_Map[I, String]
List[I]
List[List[Boolean]]

F[List, Number]
G[S, String]
```

Example 3.3.2 Given the type definitions of the previous example, the following types are malformed:

```
Tree_Map[I]               // wrong number of parameters
Tree_Map[List[I], Number] // type parameter List not within bound

F[Number, Boolean]        // Number is not a type constructor
F[Tree_Map, Number]       // Tree_Map takes two parameters,
                          //   F expects a type constructor taking one

G[S, Number]              // type S constrains its parameter to
                          //   conform to String,
                          // G expects type constructor with a parameter
                          //   that conforms to Number
```

3.3.5 Tuple Types

Syntax:

```
Simple_Type ::= '(' Types ')'
```

A tuple type $(T_1, \ldots, T_n)$ is an alias for the class `Tuple_`n`[`$T_1, \ldots, T_n$`]`, where $n \geq 2$.

Tuple classes are available as patterns for pattern matching. The properties can be accessed as methods `[1]`, …, `[`n`]` (using an "offset" that is outside of the tuple's size results in a method-not-found error, not offset-out-of-bounds – tuple classes do not implement the operator `[i]` for arbitrary i).

Tuple classes are generated lazily by the runtime as needed, so that the language does not constrain users to tuples of only limited sizes, but allows any size.

An effort will be made to introduce a simple enough syntax for variable parameterized types, if possible, until then, `Tuple_`i are the only such types.

3.3.6 Annotated Types

Syntax:

```
Annot_Type ::= {Annotation} Simple_Type
```

An annotated type $a_1 \ldots a_n T$ attaches annotations $a_1, \ldots, a_n$ to the type T.

3.3.7 Compound Types

Syntax:

```
Compound_Type ::= Annot_Type {['prepend'] 'with' Annot_Type}
                    ['with' ['refinement'] Refine_Stats]
                  | ['refinement'] Refine_Stats
Refine_Stats  ::= '{' [Refine_Stat {semi Refine_Stat}] '}'
Refine_Stat   ::= Dcl
                  | 'type' Type_Def
```

A compound type T_1 **with** ... **with** T_n **with refinement** $\{R\}$ represents values with members as given in the component types $T_1, \ldots, T_n$ and the refinement $\{R\}$. A refinement $\{R\}$ contains declarations and definitions (§5.3.5).

If no refinement is given, the type is implicitly equivalent to the same type having an empty refinement.

A compound type may also consist of just a refinement $\{R\}$ with no preceding component types – such a type has an implicit component type `Object` and describes the member values as "any value, as long as it has what the refinement requires", thus it works like an anonymous protocol.

If a compound type does not contain a concrete class type, then `Object` is implied in case the type is used as a concrete class[7].

The keyword **refinement** instructs that the following tokens will be a part of a refinement, and the construct **refinement** $\{R\}$ is called an *anonymous refinement*, being equivalent to `Object` **with refinement** $\{R\}$, although when the refinement is a part of a compound type with more elements than just the refinement itself, the `Object` type is replaced with the class type appearing in the compound type, if any.

[7]Meaning that the compound type is used as an ad-hoc (possibly anonymous) class, e.g. to create new instances of it.

The **refinement** keyword can be omitted from a compound type that consists of more elements than just the refinement. Constructs `Object` **`with`** `{`R`}` and **`refinement`** `{`R`}` are then equal.

The **refinement** keyword may also be omitted from a compound type that consists of just the refinement, but only in contexts in which a type is expected, i.e.: parameter type declaration, value or variable type declaration, return type declaration, type argument application or type parameter declaration; but never stand-alone. If used as such, the constructs **refinement** `{`R`}` and `{`R`}` are then equal.

3.3.8 Function Types

Syntax:

```
Type          ::= [Function_Args {'->' Function_Args}]
                  '->' Return_Type
Function_Args ::= Compound_Type
                | '(' [Function_Arg {',' Function_Arg}] ')'
Function_Arg  ::= [Param_Io] ['*'] [':' variable_id] Param_Type
                | '&' Function_Type
Return_Type   ::= Type | '(' ')'
```

The type `(`$T_1, \ldots, T_n$`) -> `R represents the set of function values that take arguments of types $T_1, \ldots, T_n$ and yield results of type R. In the case of exactly one argument, type T `-> `R is a shorthand for `(`T`) -> `R. Empty arguments list is indeed also possible as `-> `R, equivalent to `() -> `R.

Function types associate to the right, e.g. `(`S`) -> (`T`) -> `R is the same as `(`S`) -> ((`T`) -> `R`)`.

Function types are shorthands for class types that conform to the `Function_`i protocol – i.e. having an `apply` function or simply *being* a function. The n-ary function type `(`$T_1, \ldots, T_n$`) -> `R is a shorthand for the protocol `Function_`n`[`$T_1, \ldots, T_n, R$`]`. Such protocols are defined in the Coral library for any $n \geq 0$:

```
protocol Function_n[-$T_1 ,..., -$T_n, +$R]
  message apply (x_1 : $T_1 ,..., x_n : $T_n): $R
  ...
end protocol
```

Function types are covariant in their result type and contravariant in their argument types (§4.6).

A function return type may be declared as simple "`()`", which is equivalent to the return type of `Unit`, which in turn is equivalent to empty return type, similar to **`void`** in C-related languages. Such a type is then written equivalently as either `(`S`) -> ()` or `(`S`) -> Unit`.

Function arguments may be optionally annotated with more requirements:

- Parameter prefixed with `Param_Io` defines whether the parameter is required to be an output parameter.
- Parameter prefixed with "∗" is a requirement of a repeated parameter.
- Parameter prefixed with `:`x is a requirement of a named parameter. Note that a requirement of a parameter capturing other named parameters is not available.
- Parameter prefixed with "&" is a requirement for the passed block. It does not tell whether the function must capture the passed block, it only restricts the requirements on the particular block. The actual passed block may have more or even less positional or named parameters (extra ones on the block side are given **`nil`**, unless the type is not nullable (§5.11) – that is an error; and extra ones on the type side are simply discared), but the return type of the passed block must conform (§3.5.2).

3.3.9 Existential Types

Syntax:

```
Type                ::= Compound_Type Existential_Clauses
Existential_Clauses ::= {'for-some' '{' Existential_Dcl
                            {semi Existential_Dcl} '}'}
Existential_Dcl     ::= 'type' Type_Dcl
                      | 'val' Val_Dcl
                      | 'var' Var_Dcl
```

An existential type has the form T **`for-some`** $\{Q\}$, where Q is a sequence of type declarations. Let $t_1[tps_1]$ `>:` L_1 `<:` U_1 , ..., $t_n[tps_n]$ `>:` L_n `<:` U_n be the types declared in Q.

A *type instance* of T **`for-some`** $\{Q\}$ is a type σT, where σ is a substitution over $t_1, \ldots, t_n$, such that for each i, L_i `<:` t_i `<:` U_i. The set of values denoted by the existential type T **`for-some`** $\{Q\}$ is the union of the set of values of all its type instances.

3.3.10 Nullable Types

Syntax:

```
Nullable_Type ::= Type [Nullable_Mod]
Nullable_Mod  ::= '?' | '!'
```

A nullable type has the form T`?` or T`!`, where "`?`" denotes explicitly a nullable type, and "`!`" denotes explicitly not-nullable type. Although **`nil`** as the singleton member

of the `Nothing` type is a subtype of every type, Coral types are implicitly not-nullable, meaning it's not possible to pass **`nil`** where an instance of T is expected, unless T is of course `Nothing`. Nullability (§5.11) is one of the intrinsic properties of every class type.

Explicitly nullable types are handled by an intrinsic anonymous subtype of T, which is explicitly nullable, overriding the preference of T. Explicitly not-nullable types are handled by an intrinsic anonymous subtype of T, which is explicitly not-nullable, overriding the preference of T. Explicit nullability of already nullable types is redundant, as is explicit non-nullability of already not-nullable types. Explicit nullability of the `Option` type is also redundant and is in fact ignored.

Nullable types in this form can appear as types of variables, parameters and return types.

3.3.11 Dependent Types

Syntax:

```
Simple_Type ::= Simple_Type Dep_Args
Dep_Args    ::= '@[' Dep_Arg {',' Dep_Arg} ']'
Dep_Arg     ::= variable_id | Dep_Sort_Val
```

Dependent types in Coral are implemented by simple indexed types. Dependent types are explained in greater detail in §5.8.

Example 3.3.3 The following are examples of dependent types representing the number 42 and all strings of length 42:

```
Integer@[42]
String@[42]
```

3.4 Non-Value Types

The types explained in the following paragraphs do not appear explicitly in programs, they are internal and do not represent any type of value directly.

3.4.1 Method Types

A method type is denoted internally as $(Ps) \mapsto R$, where (Ps) is a sequence of parameter names, types and extra properties $(ep_1 : T_1, \ldots, ep_n : T_n)$ for some $n \geq 0$ and R is a (value or method) type. This type represents named or anonymous methods that take arguments named $p_1, \ldots, p_n$ of types $T_1, \ldots, T_n$, have extra properties e and return

a result of type R. Names of parameters are either simple identifiers (for positional argument passing) or symbol literals (§1.5.8, for named arguments passing – they make difference between method types with possibly same parameter types, therefore the name is a part of the method type along with the associated parameter type[8]).

Method types associate to the right:[9]
$(Ps_1) \mapsto (Ps_2) \mapsto R$ is treated as $(Ps_1) \mapsto ((Ps_2) \mapsto R)$.

A special case are types of methods without any parameters. They are written here as $() \mapsto R$.

Another special case are types of methods without any return type. They are written here as $(Ps) \mapsto ()$. Methods that have this return type do not have an implicit return expressions and an attempt to return a value from it results in a compile-time error.[10]

Method types do not exist as types of values. If a method name is used as a value, its type is implicitly converted to a corresponding function type (§3.3.8).

Extra properties of parameters are as follows: a $*$ for variable arguments, $**$ for any named arguments and & for a captured block argument, or nothing for regular parameters.

Example 3.4.1 The declarations

```
def a: -> Integer // or def a () -> Integer
def b (x : Integer): Boolean
def c (x : Integer): (y : String, z : String) -> String
def d (:x : Integer): Integer
def e (*x : Integer): Integer
def f (Integer): ()
def g (Integer)(Integer): Integer
def h (Integer): (Integer) -> Integer
```

produce the typings

```
a : () ↦ Integer
b : (Integer) ↦ Boolean
c : (Integer) ↦ (String, String) ↦ String
d : (:x Integer) ↦ Integer
e : (*Integer) ↦ Integer
f : (Integer) ↦ ()
g : (Integer) ↦ (Integer) ↦ Integer
g : (Integer) ↦ (Integer) ↦ Integer
```

[8]This means that, for simplicity, if we have a method with one parameter, which is a named parameter, represented by having its name expressed with a symbol literal, and the parameters have an equivalent type, but different names, the method types are not equivalent.

[9]Like in Haskell or Scala.

[10]A compile-time error like this may happen during a runtime evaluation as well.

The difference between the "g" and "h" functions is that using the chain of return types as in function "g", the function body is automatically curried to return a function that is of type `(Integer) ↦ Integer`. With the function "h", currying has to be implemented manually.

3.4.2 Polymorphic Method Types

A polymorphic method type is the same as a regular method type, but enhanced with a type parameters section. It is denoted internally as $[tps] \mapsto T$, where $[tps]$ is a type parameter section $[a_1 >: L_1 <: U_1 , \ldots, a_n >: L_n <: U_n]$ for some $n \geq 0$ and T is a (value or method) type. This type represents (only[11]) named methods that take type arguments $S_1, \ldots, S_n$, for which the lower bounds $L_1, \ldots, L_n$ conform (§3.5.2) to the type arguments and the type arguments conform and the upper bounds $U_1, \ldots, U_n$ and that yield results of type T. No explicit lower bound implies `Nothing` to be the corresponding lower bound, no explicit upper bound implies `Object` to be the corresponding upper bound. As usual, lower bound must conform to the corresponding upper bound.

Example 3.4.2 The declarations

```
def empty[$A]: List[$A]
def union[$A <: Comparable[$A]] (x : Set[$A],
         xs : Set[$A]): Set[$A]
```

produce the typings

```
empty : [$A >: Nothing <: Object] () ↦ List[$A]
union : [$A >: Nothing <: Comparable[$A]] (Set[$A],
        Set[$A]) ↦ Set[$A]
```

3.4.3 Type Constructors

A type constructor is in turn represented internally much like a polymorphic method type. $[\pm a_1 >: L_1 <: U_1, \ldots, \pm a_n >: L_n <: U_n] \mapsto T$ represents a type that is expected by a type constructor parameter. The difference is that the represented internal entity is not a method, but a type, creating higher-kinded types.

3.5 Relations Between Types

We define two relations between types.

[11] Not anonymous.

Type equivalence	$T \equiv U$	T and U are interchangeable in all contexts.
Conformance	$T <: U$	Type T conforms to type U.

3.5.1 Type Equivalence

Equivalence (≡) between types is the smallest congruence, such that the following statements are true:

- If t is defined by a type alias **`type t is T`**, then t is equivalent to T.
- If a path p has a singleton type q**`#singleton-type`**, then p**`#singleton-type`** ≡ q**`#singleton-type`**.
- Two compound types (§3.3.7) are equivalent, if the sequences of their components are pairwise equivalent, occur in the same order and their refinements are equivalent.
- Two refinements (§3.3.7 & TBD: named refinements) are equivalent, if they bind the same names and the modifiers, types and bounds of every declared entity are equivalent in both refinements. Two equivalent refinements, both or one attached to a compound type, do not imply the compound types to be equivalent. This applies to both anonymous and named refinements.
- Two method types (§3.4.1) are equivalent, if they have equivalent return types, both have the same number of parameters and corresponding parameters have equivalent types and extra properties. Names of parameters matter for method type equivalence only with named parameters.
- Two polymorphic method types (§3.4.2) are equivalent, if they have the same number of type parameters, the return types are equivalent as well as lower and upper bounds of corresponding type parameters.
- Two existential types (§3.3.9) are equivalent, if they have the same number of quantifiers and the quantified types as well as lower and upper bounds of corresponding quantifiers are equivalent.
- Two type constructors (§3.4.3) are equivalent, if they have the same number of type parameters, the return types are equivalent as well as variances, lower and upper bounds of corresponding type parameters.

3.5.2 Conformance

The conformance relation (<:) is the smallest transitive relation that satisfies the following conditions:

- Conformance includes equivalence, therefore if $T \equiv U$, then $T <: U$.
- For every value type T, `Nothing` $<: T <:$ `Object`.
- For every type constructor T with any number of type parameters, `Nothing` $<: T <:$ `Object`.
- A type variable t conforms to its upper bound and its lower bound conforms to t.
- A class type or a parameterized type conforms to any of its base types.

3.5.3 Weak Conformance

For now, *weak conformance* is a relation defined on members of the `Number` type as a relaxation of conformance. The relation is simple: a type t weakly conforms to another type u when u's size contains all values of t (we say that t can be converted to u without precision loss).

Chapter 4

Basic Declarations & Definitions

Syntax:

```
Dcl         ::= 'val' Val_Dcl
              | 'var' Var_Dcl
              | 'def' Def_Dcl
              | 'message' Fun_Dcl 'end' ['message']
              | 'function' Fun_Dcl 'end' ['function']
              | 'type' Type_Dcl
Pat_Var_Def ::= 'val' Pat_Def
              | 'var' Pat_Def
              | 'let' ['!'] variable_id ':=' Expr
Def         ::= Pat_Var_Def
              | 'def' Fun_Def
              | 'method' Fun_Def 'end' ['method']
              | 'method' Fun_Alt_Def
              | 'function' Fun_Def 'end' ['function']
              | 'function' Fun_Alt_Def
              | 'type' Type_Def
              | Tmpl_Def
```

A *declaration* introduces names and assigns them types. Using another words, declarations are abstract members, working sort of like header files in C.

A *definition* introduces names that denote terms or types. Definitions are the implementations of declarations.

Both declarations and definitions produce *bindings* that associate type names with type definitions or bounds, and that associate term names with types.

Even more simply put, declarations declare a binding with a type (or type-less), and definition defines the term behind that binding (along with the binding).

4.1 Value Declarations & Definitions

Syntax:

```
Dcl         ::= 'val' Val_Dcl
Val_Dcl     ::= var_ids ':' Type
Pat_Var_Def ::= 'val' Pat_Def
              | 'let' ['!'] variable_id ':=' Expr
Pat_Def     ::= Pattern2 {',' Pattern2} [':' Type] ':=' Expr
              | [[var_ids ',']
                '*' variable_id] var_ids
                ':=' [[Expr {',' Expr} ','] '*' Expr]
                Expr {',' Expr}
var_ids     ::= var_id {',' var_id}
var_id      ::= variable_id | ivar_id | cvar_id
```

A value declaration **`val`** x`:` T introduces x as a name of a value of type T. May appear in any block of code and an attempt to use it prior to initialisation with a value is an error. More specifically, a value declaration **`val`** `@`x`:` T introduces x as a name of an instance value of type T, and a value declaration **`val`** `@@`x`:` T introduces x as a name of a class instance value of type T.

A value definition **`val`** x`:` T `:=` e defines x as a name of the value that results from evaluation of expression e.

A value in this sense[1] is an immutable variable. A declared value can be assigned just once[2], a defined value is already assigned from its definition.

The value type T may be always omitted, in that case the type is inferred and bound to the name. If a type T is omitted, the type of expression e is assumed. If a type T is given, then e is expected to conform to it (§3.5.2).

Evaluation of the value definition implies evaluation of its right-hand side e, unless it has a modifier **`lazy`** – in that case, evaluation is deferred to the first time the value is accessed.

A *lazy value* is of the form

`lazy val` x`:` T `:=` e

A lazy value may only be defined, and a value of the same name (binding) may be declared prior to the value definition, but never as a lazy value.

[1] Everything in Coral is a value – remember, Coral is also a functional language, to some extent.

[2] A similar way that **`final`** variables or members in Java can be assigned just once, but Java furthermore requires that this assignment will happen in every code path, Coral does not impose such requirement.

The effect of the value definition is to bind x to the value of e converted to type T.

A *constant value definition* is of the form

```
let x: T := e
```

where e is an expression that is supposed to be treated as constant in the same block from its occurrence on. Values defined with **let** have certain limitations and properties:

1. They can't use patterns instead of a name.
2. They can't be lazy.
3. They can't be used in a declaration, only in a definition.
4. They can be used to redefine a variable (the name is then treated as a new binding in the scope).
5. They can't define (class) instance variables.
6. They can be used in workflows (§6.27).[3]

The type T may be omitted.

Value declarations & definitions with the type T omitted are of the form

```
val x
val @x
val @@x
val x := e
val @x := e
val @@x := e
let x := e
```

A value declaration without any type is basically only declaring the name, so that a binding is introduced and the actual value is for another code to define.[4]

A value definition can alternatively have a pattern (§8.1) as left-hand side (the name). If p is a pattern other than a simple name or a name followed by a colon and a type, then the value definition **val** p := e is expanded as follows:

[3]A pragma that would turn all values into lazy values might exist, and lazy values should never appear in workflows, so that's why **val** should not be allowed in workflows.

[4]Usually, that another code should be a **constructor** or the class-level block in another file, maybe.

1. If the pattern p has bound variables $x_1, \ldots, x_n$ for some $n > 1$:

```
val x$ := match e
  when p then (x_1, ..., x_n)
end match
val x_1 := x$[1]
...
val x_n := x$[n]
```

2. If p has exactly one unique bound variable x:

```
val x := match e
  when p then x
end match
```

3. If p has no bound variables:

```
match e
  when p then ()
end match
```

Example 4.1.1 The following are examples of value definitions.

```
val pi := 3.14159
val pi: Double := 3.14159
val Some(x) := f()
val Some(x), y := f()
val x ~> xs := my_list
```

The last three definitions have the following expansions:

```
val x := match f()
  when Some(x) then x
end match

val x$ = f()
val x := match x$
  when Some(x) then x
end match
val y := x$

val x$ := match my_list
  when x ~> xs then (x, xs)
end match
val x := x$[1]
val xs := x$[2]
```

The name of any declared or defined value must not end with "_=".

The following shorthands are recognized:

A value declaration **val** $x_1, \ldots, x_n$**:** T is a shorthand for the sequence of value declarations **val** x_1**:** T**;** ...**;** **val** x_n**:** T.

A value definition **val** $p_1, \ldots, p_n$ **:=** e is a shorthand for the sequence of value definitions **val** p_1 **:=** e**;** ...**;** **val** p_n **:=** e.

A value definition **val** $p_1, \ldots, p_n : T$ **:=** e is a shorthand for the sequence of value definitions **val** $p_1 : T$ **:=** e**;** ...**;** **val** $p_n : T$ **:=** e.

4.2 Variable Declarations & Definitions

Syntax:

```
Dcl ::= 'var' Var_Dcl
Pat_Var_Def ::= 'var' Var_Def
Var_Dcl ::= var_ids ':' Type
```

A variable declaration **var** x**:** T introduces a mutable variable without a defined initial value of type T. More specifically, **var** `@`x**:** T introduces a mutable instance variable of type T and **var** `@@`x**:** T introduces a mutable class instance variable of type T.

A variable definition **var** x**:** T **:=** e defines x as a name of the value that results from evaluation of expression e. The type T can be omitted, in that case the type of expression e is assumed, but not bound to the variable – the variable is only bound to `Object` then. If the type T is given, then e is expected to conform to it (§3.5.2), as well as every future value of the variable.

Variable definitions can alternatively have a pattern (§8.1) as their left-hand side. A variable definition **var** p **:=** e, where p is a pattern other than a simple name followed by a colon and a type, is expanded in the same way (§4.1) as a value definition **val** p **:=** e, except that the free names in p are introduced as mutable variables instead of values.

The name of any declared or defined variable must not end with "_=".

The following shorthands are recognized:

A variable declaration **var** $x_1, \ldots, x_n$**:** T is a shorthand for the sequence of variable declarations **var** x_1**:** T**;** ...**;** **var** x_n**:** T.

A variable definition **var** $x_1, \ldots, x_n$ **:=** e is a shorthand for the sequence of variable definitions **var** x_1 **:=** e**;** ...**;** **var** x_n **:=** e.

A variable definition **var** $x_1, \ldots, x_n$**:** T **:=** e is a shorthand for the sequence of variable definitions **var** x_1**:** T **:=** e**;** ...**; var** x_n**:** T **:=** e.

4.3 Property Declarations & Definitions

Syntax:

```
Prop_Dcl   ::= 'property' ['(' Prop_Specs ')'] simple_id
               [':' Type]
Prop_Specs ::= Prop_Spec {',' Prop_Spec}
Prop_Spec  ::= ([Access_Modifier] ('get' | 'set')) | 'weak'
Prop_Def   ::= 'property' ['(' Prop_Specs ')'] simple_id
               [':' Type]
               '{' Prop_Impl {semi Prop_Impl} '}'
Prop_Impl  ::= ('get' [Prop_Get_Impl])
               | ('set' [Prop_Set_Impl])
               | ('val' ':=' Expr)
               | ('var' ':=' Expr)
```

A property declaration **property** x**:** T introduces a property without a defined initial value of type T. Property declaration does not specify any actual implementation details of how or where the declared value is stored.

A property definition **property** x**:** T {**get** ...**; set** ...} introduces a property with a possibly defined initial value of type T. Property definition may specify implementation details of the behavior and storage of a property, but may as well opt-in for auto-generated implementation, which is:

1. Storage of the property's value is in an instance variable (or a class instance variable in case of class properties) of the same name as is the name of the property: **property** x is stored in an instance variable `@`x.
2. Properties defined with only **get** are stored in immutable instance variables (§4.1).
3. Properties defined with **set**[5] are stored in mutable instance variables (§4.2).
4. Properties defined with **weak** are stored as weak references. A property **property** x**:** T is stored in an instance of type `Weak_Reference[`T`]`.

Declaring a property x of type T is equivalent to declarations of a *getter function* x and a *setter function* x`_=`, declared as follows:

[5]It is also possible to declare/define properties that are **set**-only. That makes them *write-only*, as opposed to *read-only* properties with **get**-only.

```
def x (): T; end
def x_= (y: T): (); end
```

Assignment to properties is translated automatically into a setter function call and reading of properties does not need any translation.

4.4 Type Declarations & Aliases

Syntax:

```
Dcl      ::= 'type' Type_Dcl 'end' ['type']
Type_Dcl ::= constant_id [Type_Param_Clause] [Dep_Params]
             [UoM_Params] ['>:' Type] ['<:' Type]
Def      ::= 'type' Type_Def
Type_Def ::= constant_id [Type_Param_Clause] [Dep_Params]
             [UoM_Params] ':=' Type
```

A *type declaration* **type** *t*`[`*tps*`]@[`*dps*`][<`*uomps*`>] >:` *L* `<:` *U* declares *t* to be an abstract type with lower bound type *L* and upper bound type *U*. If the type parameter clause `[`*tps*`]` is omitted, *t* abstracts over a first-order type, otherwise *t* stands for a type constructor that accepts type arguments as described by the type parameter clause. Omitting indexing parameter clause `@[`*dps*`]` or units of measure parameter clause `[<`*uomps*`>]` does not affect whether *t* abstracts over a first-order type, since these parameters only restrict subsets of the type *t*, instead of constructing new types.

If a type declaration appears as a member declaration of a type, implementations of the type may implement *t* with any type *T*, for which $L <: T <: U$. It is an error if *L* does not conform to *U*. Either or both bounds may be omitted. If the lower bound *L* is omitted, the bottom type `Nothing` is implied. If the upper bound *U* is omitted, the top type `Object` is implied.

A type constructor declaration declaration imposes additional restriction on the concrete types for which *t* may stand. Besides the bounds *L* and *U*, the type parameter clause, indexing parameter clause and units of measure parameter clause may impose higher-order bounds and variances, as governed by the conformance of type constructors (§3.5.2).

The scope of a type parameter extends over the bounds `>:` *L* `<:` *U* and the type parameter clause *tps* itself. A higher-order type parametre clause (of an abstract type constructor *tc*) has the same kind of scope, restricted to the declaration of the type parameter *tc*.

To illustrate nested scoping, these declarations are all equivalent:

```
type t[:m[$x] <: Bound[$x], Bound[$x]] end
```

```
type t[:m[$x] <: Bound[$x], Bound[$y]] end

type t[:m[$x] <: Bound[$x], Bound[_]] end,
```

as the scope of, e.g., the type parameter of m is limited to the declaration of m. In all of them, t is an abstract type member that abstracts over two type constructors: m stands for a type constructor that takes one type parameter and that must be a subtype of `Bound`, t's second type constructor parameter. However, the first example should be avoided, as the last `$x` is unrelated to the first two occurrences, but may confuse the reader.

A *type alias* **type** t `:=` T defines t to be an alias name for the type T. Since—for type safety and consistence reasons—types are constant and can not be replaced by another type when bound to a constant name, type aliases are permanent. A type remembers the first given constant name, no alias can change that. The left hand side of a type alias may have a type parameter clause, e.g. **type** t[*tps*] `:=` T. The scope of a type parameter extends over to the right hand side T and the type parameter clause *tps* itself.

It is an error if a type alias refers recursively to the defined type constructor itself.

Example 4.4.1 The following are legal type declarations and aliases:

```
type Integer_List := List[Integer]
type T <: Comparable[T]
type Two[$A] := Tuple_2[$A, $A]
type My_Collection[+$X] <: Iterable[$X]
```

The following are illegal:

```
type Abs := Comparable[Abs] // recursive type alias

type S <: T  // S, T are bounded by themselves
type T <: S

type T >: Comparable[T.That] // can't select from T
                             // T is a type, not a value
type My_Collection <: Iterable
  // type constructor members must explicitly state
  // their type parameters
```

4.5 Type Parameters

Syntax:

```
Type_Param_Clause  ::= '[' Variant_Type_Param
                       {',' Variant_Type_Param} ']'
Variant_Type_Param ::= {Annotation} ['+' | '-'] Type_Param
Type_Param         ::= (tp_id | '_') [Type_Param_Clause]
                       ['>:' Type] ['<:' Type]
```

Type parameters appear in type definitions, class definitions and function definitions.

The most general form of a first-order type parameter is `@{`a_1`}`...`@{`a_n`}` $\pm$ t `>:` L `<:` U. L is a lower bound and U is an upper bound. These bounds constrain possible type arguments for the parameter. It is an error if L does not conform to U.[6] Then, $\pm$ is a *variance* (§4.6), i.e. an optional prefix of either + or −. The type parameter may be preceded by one or more annotation applications (§6.14 & §10).

Example 4.5.1 The following are some well-formed type parameter clauses:

```
[$S, $T]
[@{Specialized} $S, $T]
[$Ex <: Raisable]
[$A <: Comparable[$B], $B <: $A]
```

4.6 Variance of Type Parameters

Variance annotations indicate how instances of parameterised types relate with respect to subtyping (§3.5.2). A "+" variance indicates a covariant dependency, a "−" variance indicates a contravariant dependency, and an empty variance indicates an invariant dependency.

A variance annotation constrains the way the annotated type variable may appear in the type or class which binds the type parameter.

4.7 Function Declarations & Definitions

Syntax:

```
Dcl          ::= 'def' Fun_Dcl 'end' ['def']
               | 'message' Fun_Dcl 'end' ['message']
               | 'function' Fun_Dcl 'end' ['function']
Fun_Dcl      ::= Fun_Sig ':' Type
Def          ::= 'def' Fun_Def 'end' ['def']
```

[6] This is sometimes detectable as soon as during compilation.

```
                | 'def' Fun_Alt_Def
                | 'method' Fun_Def 'end' ['method']
                | 'method' Fun_Alt_Def
                | 'function' Fun_Def 'end' ['function']
                | 'function' Fun_Alt_Def
Fun_Def       ::= Fun_Sig [':' Return_Type] [Fun_Dec] [semi Fun_Stats]
Fun_Alt_Def   ::= Fun_Sig [':' Return_Type] ':=' Expr
Fun_Dec       ::= [semi] 'declare' Fun_Dec_Exprs [semi] 'begin'
Fun_Dec_Exprs ::= Fun_Dec_Expr {semi Fun_Dec_Expr}
Fun_Sig       ::= Function_Path [Fun_Tpc] Param_Clauses
Fun_Tpc       ::= '[' Type_Param {',' Type_Param} ']'
Function_Path ::= function_id
                | 'self' '.' function_id
                | variable_id '.' function_id
Param_Clauses ::= {Param_Clause} ['(' 'implicit' Params ')']
Param_Clause  ::= '(' [Params] ')'
Params        ::= Param {',' Param}
Param         ::= {Annotation} [Param_Extra] variable_id
                  [':' Param_Type] [':=' Expr]
                  | Literal
Param_Extra   ::= [Param_Io] [Param_Rw] ['*' | '**' | '&'] [':']
Param_Io      ::= 'in' ['out'] | 'out'
Param_Rw      ::= 'val' | 'var'
Param_Type    ::= Type | '$' constant_id | '=>' [Type | '_']
```

A function declaration has the form of **def** f $psig$`:` T, where f is the function's name, $psig$ is its parameter signature and T is its return type.

A function definition **def** f $psig$`:` T `:=` e also includes a *function body* e, i.e. an expression which defines the functions's return value. A parameter signature consists of an optional type parameter clause `[`tps`]`, followed by zero or more value parameter clauses `(`ps_1`)`...`(`ps_n`)`. Such a declaration or definition introduces a value with a (possibly polymorphic) method type, whose parameter types and return types are as given.

Multiple parameter clauses render curried functions.

The type of the function body is expected to conform (§3.5.2) to the function's declared result type, if one is given.

If the function's result type is given as one of "`()`" or `Unit`, the function's implicit return value is stripped and it is an error if a return statement occurs in the function body with a value to be returned, unless the return value is specified again as "`()`".

An optional type parameter clause tps introduces one or more type parameters, possibly with bounds. The scope of a type parameter includes the whole signature, including any of the type parameter bounds as well as the function body, if present.

A value parameter clause ps consists of zero or more formal parameter bindings, such as x`:` T or x`:` T `:=` e, which bind value parameters and associate them with their types. Each value parameter declaration may optionally define a default value expression. The value expression is represented internally by an invisible function, which gets called when the function matched the function call and an explicit value for the parameter was not provided.

The order in which different kinds of value parameters may appear is as follows:

1. n mandatory positional parameters (§4.7.4): x`:` T, where $n \geq 0$.
2. n optional positional parameters (§4.7.5): x`:` T `:=` e, where $n \geq 0$.
3. n repeated parameters (§4.7.6): $*x$`:` T, where $0 \leq n \leq 1$.
4. n post mandatory positional parameters (§4.7.4): x`:` T, where $n \geq 0$.
5. n named parameters (§4.7.7): `:`x `:` T or `:`x `:` T `:=` e, where $n \geq 0$.
6. n captured named parameters (§4.7.7): $**x$`:` T, where $0 \leq n \leq 1$.
7. n captured block parameters (§4.7.8): `&`x`:` T, where $0 \leq n \leq 1$.

For every parameter $p_{i,j}$ with a default value expression, a function named

`default$`n

is generated inside the function, inaccessible for user programs. Here, n denotes the parameter's position in the method declaration. These methods are parameterized by the type parameter clause `[`tps`]` and all value parameter clauses `(`ps_1`)`...`(`ps_{i-1}`)` preceeding $p_{i,j}$.

The scope of a formal value parameter name x comprises all subsequent parameter clauses, as well as the method return type and the function body, if they are given.

Example 4.7.1 In the method

```
def compare[$T](a: $T := 0)(b: $T := a) := (a = b)
```

the default expression `0` is type-checked with an undefined expected type. When applying `compare()`[7], the default value `0` is inserted and `T` is instantiated to `Number`. The functions computing the default arguments have the forms[8]:

```
def default$1[$T]: Number := 0
def default$2[$T](a: $T): $T := a
```

[7]Without any explicit arguments.

[8]See, at the moment `default$2` is called, the parameter `a` is already computed and passed as an argument to it.

Parameters may be optionally flagged with **`var`** and **`val`** keywords, modifying their mutability inside the method. Some combinations are disallowed, as explained in the following sections. All parameters are implicitly **`val`**-flagged, unless the parameter kind implies a **`var`** flag, such as an output parameter (§4.7.3).

4.7.1 By-Name Parameters

Syntax:

```
Param_Type ::= '=>' [Type | '_']
```

The type of a value parameter may be prefixed by "=>", e.g. x`: =>` T. This indicates that the corresponding argument is not evaluated at the point of function application, but instead is evaluated at each use within the function. That is, the argument is evaluated using *call-by-name*.

The by-name modifier is disallowed for output parameters (§4.7.3 & §6.6.3) and for implicit parameters (§7.2). The by-name modifier implies **`val`** parameter and is disallowed for **`var`** parameters.

By-name parameters require a specific application (§6.6.2). A by-name parameter bound to a wildcard type "_" matches any type of by-name argument.

By-name parameters with default value expressions evaluate the default value expression each time the parameter is accessed, unlike optional parameters that evaluate the default value expression only once.

By-name parameters imply the **`val`** flag, and disallow the **`var`** flag.

4.7.2 Explicit Parameters

Syntax:

```
Param ::= Literal
```

The parameter may be specified by its literal value. Such parameters may only appear where positional parameters may appear. The type of the parameter is the type of the literal value. Methods with explicit parameters are preferred during method resolution to methods with the same parameter types (§6.6), but it is an error if more than one method with explicit parameters match the function application.

The recommendation for usage of these parameters are:

- Use explicit parameters with unary methods only.
- If the value is a collection, use an empty collection literal only.

Example 4.7.2 Sample methods that use explicit parameters:

```
def factorial (0) := 1
def factorial (x) := x * factorial(x - 1)
```

Since the parameter has no name to bind to, it is not accessible inside the method body.

4.7.3 Input & Output Parameters

Syntax:

```
Param_Io ::= 'in' ['out'] | 'out'
```

If no input/output parameter specifier is explicitly available, then the parameter is implicitly an input parameter. Output parameters require a specific application (§6.6.3).

Output parameters imply **var** parameter and is disallowed for **val** parameters. Input parameters that are not output parameters at the same time can be both **var** and **val**.

4.7.4 Positional Parameters

Positional parameters are of the forms:

```
x: T
var x: T
val x: T
in x: T
in var x: T
in val x: T
out x: T
out var x: T
in out x: T
in out var x: T
x: => T
val x: => T
in x: => T
in val x: => T
```

Positional parameters may not have any modifiers, except for input/output modifiers (§4.7.3) and by-name (§4.7.1). Positional parameters can't have any default value expressions.

4.7.5 Optional Parameters

Optional parameters are of the forms:

```
x: T := e
var x: T := e
val x: T := e
in x: T := e
in var x: T := e
in val x: T := e
x: => T := e
val x: => T := e
in x: => T := e
in val x: => T := e
```

Optional parameters may not have any modifiers, except for input/output modifiers (§4.7.3)[9] and by-name modifier (§4.7.1). Optional parameters have a *default value expressions* and may appear between positional parameters, being followed by any number of positional parameters (including no more positional parameters at all), or being followed by repeated parameters and then positional parameters (§6.6.1). Optional parameters are disallowed for output parameters and repeated parameters.

4.7.6 Repeated Parameters

Repeated parameters are of the forms:

```
*x: T
var *x: T
val *x: T
in *x: T
in var *x: T
in val *x: T
```

Between optional parameters and the tailing positional parameters may be a value parameter prefixed by "*", e.g. `(..., *x: T)`. The type of such a *repeated* parameter inside the method is then a list type `List[T]`. Methods with repeated parameters take a variable number of arguments of type T between the optional parameters block and the last positional parameters block, including no arguments at all (an empty list is then its value).

If a repeated parameter is flagged with **`val`**, the parameter itself is immutable, not the elements of the list. Repeated parameters are **`val`**-flagged implicitly, unless explicitly flagged as **`var`**, to protect the captured elements from accidental overwrite.

Example 4.7.3 The following method definition computes the sum of the squares of a variable number of integer arguments.

[9] Optional parameters are always "input", so that declaration is always redundant.

```
def sum (*args: Integer): Integer
declare
  var result := 0
begin
  for arg in args loop
    result += arg ** 2
  end loop

  result
end
```

The following applications of this method yield `0`, `1`, `14`, in that order.

```
sum
sum 1
sum 1, 2, 3
```

Furthermore, assume the definition:

```
val xs := %[1, 2, 3]
```

The following application of the method `sum` is not resolved:[10]

```
sum xs // Error: method match not found, wrong arguments
```

By contrast, the following application is well-formed and yields again the result `14`:

```
sum *xs
```

4.7.7 Named Parameters

Named parameters are of the forms:

```
:x : T
var :x : T
val :x : T
in :x : T
in var :x : T
in val :x : T
out :x : T
out var :x : T
in out :x : T
in out var :x : T
:x : T := e
var :x : T := e
```

[10]Unless there is an overloaded version of the method that accepts a list of integers as its parameter.

```
val :x : T := e
in :x : T := e
in var :x : T := e
in val :x : T := e
:x : => T
val :x : => T
in :x : => T
in val :x : => T
:x : => T := e
val :x : => T := e
in :x : => T := e
in val :x : => T := e
```

Captured named parameters are of the form:

```
**x : T
var **x : T
val **x : T
in **x : T
in var **x : T
in val **x : T
```

Named parameters are a way of allowing users of the method to write down arguments in any order, provided that their name is given at function application (§6.6 & §6.6.1). Named parameters may have a default value expression and may be both input and output. Named parameters are disallowed for repeated parameters. Named parameters inside the method is then accessible the same way as a positional parameter.

Captured named parameters are capturing any other applied named parameters that were not captured by their explicit declaration (§6.6.1). They are declared after the block of named parameters, prefixed by "**", e.g. `(..., **x: T)`. The type of such a captured named parameter inside the method is then a dictionary type `Dictionary[Symbol, T]`. Methods with captured named parameters take a variable number of named arguments of type T mixed with other named arguments and before the captured block parameter. Captured named parameters are disallowed for repeated parameters, output parameters and by-name parameters.

If a captured named parameter is flagged with **`val`**, the parameter itself is immutable, not the elements of the dictionary. Captured named parameters are **`val`**-flagged implicitly, unless explicitly flagged as **`var`**, to protect the captured elements from accidental overwrite.

4.7.8 Captured Block Parameter

Captured block parameters are of the forms:

```
&x
&x: T
in &x
in var &x
in val &x
in &x: T
in var &x: T
in val &x: T
```

Captured block parameter is a way to capture an applied block that is otherwise passed in implicitly as a function into **yield** expressions . The forms of captured block parameters explicitly denote the case without the block's function type, since block parameters receive any arguments and those missing are implicitly set to **nil**. The function type of the block may be used to further constrain the applied block argument, but is not used during method resolution (§6.6).

It is an error if a block parameter type T is provided and it is not a function type (§3.3.8). It is also an error if the applied block argument does not accept the arguments declared by the type T, or if the block would not return a value conforming to the return type required by the type T. The applied block argument may accept more arguments than required by T, however, these will be set implicitly to **nil**. Also, the applied block argument may itself require less constrained parameter types, in which case the arguments applied to it must (and will) always conform (§3.5.2) to the block's parameter requirements. Whether the function type T has a return type or not is irrelevant.

If a block parameter type T is given, then the applied block argument must accept parameters, such that the parameter constrains declared by the function type T conform to the parameter constrains declared by the applied block: the parameters of the applied block must be the same or less restrictive than those declared by T – they must be pairwise contravariant or invariant, never covariant (§4.6).

4.7.9 Parameter Kind Combinations

This section is *normative*.

Users should design their functions in a way that makes them having as few parameters as possible.

In this spirit, an ideal count of parameters lies between 0 and 2 (nullary, unary and binary functions). Named parameters should be used with only up to one positional[11] parameter. Functions with more than 4 different parameters should be avoided.

[11] In this sense, optional and repeated parameters are also positional, because they are applied on a particular numbered position rather than named.

4.7.10 Method Types Inference

Parameter Type Inference. Functions that are members of a class C may define parameters without type annotations. The types of such parameters are inferred as follows. Say, a method m in a class C has a parameter p which does not have a type annotation. We first determine methods m' in C that might be overridden (§5.1.6) by m, assuming that appropriate types are assigned to all parameters of m whose types are missing. If there is exactly one such method, the type of the parameter corresponding to p in that method—seen as a member of C—is assigned to p. It is an error if there are several such overridden methods m'. If there is none[12] (m does not override any m' known at compile-time), then the parameters are inferred to be of type `Object`.

Example 4.7.4 Assume the following definitions:

```
protocol I[$A]
  def f(x: $A)(y: $A): $A
end
class C
  implements I[Integer]
  def f(x)(y) := x + y
end
```

Here, the parameter and return types of `f` in `C` are inferred from the corresponding types of `f` in `I`. The signature of `f` in `C` is thus inferred to be

```
  def f(x: Integer)(y: Integer): Integer
```

Return Type Inference. A class member definition m that overrides some other function m' in a base class of C may leave out the return type, even if it is recursive. In this case, the return type R' of the overridden function m'—seen as a member of C—is taken as the return type of m for each recursive invocation of m. That way, a type R for the right-hand side of m can be determined, which is then taken as the return type of m. Note that R may be different from R', as long as R conforms to R'. If m does not override any m', then its return type is inferred to be of type `Object`.

Example 4.7.5 Assume the following definitions:

```
protocol I
  def factorial(x: Integer): Integer
end
class C
  implements I
  def factorial(x: Integer) := {
```

[12]Detected at compile-time. Dynamically added overridden methods are not used with type inference.

```
    if x = 0 then 1 else x * factorial(x - 1) end
  }
end
```

Here, it is ok to leave out the return type of `factorial` in `C`, even though the method is recursive.

For any index i let $fsig_i$ be a function signature consisting of a function name, an optional type parameter section, and zero or more parameter sections. Then a function declaration **def** $fsig_1, \ldots, fsig_n$**:** T is a shorthand for the sequence of function declarations **def** $fsig_1$**:** T**;** $\ldots$**; def** $fsig_n$**:** T. A function definition **def** $fsig_1, \ldots, fsig_n$ **:=** e is a shorthand for the sequence of function definitions **def** $fsig_1$ **:=** e**;** $\ldots$**; def** $fsig_n$ **:=** e. A function definition **def** $fsig_1, \ldots, fsig_n : T$ **=** e is a shorthand for the sequence of function definitions **def** $fsig_1 : T$ **:=** e**;** $\ldots$**; def** $fsig_n : T$ **:=** e.

4.8 Overloaded Declarations & Definitions

An overloaded definition is a set of $n > 1$ function definitions in the same statement sequence that define the same name, binding it to types $T_1, \ldots, T_n$, respectively. The individual definitions are called *alternatives*. Overloaded definitions may only appear in the expression sequence of a class-level block. Alternatives always need not to specify the type of the defined entity completely.

Overloaded function definitions have strong impact on method resolution. It is an error if a single set of arguments may be applied type-safely to multiple overloaded functions – to resolve this, explicit argument types have to be applied (§6.6).

Overloaded functions generate new functions that internally merge the overloaded functions into one, which then resolves the correct overloaded function based on the applied types.

Example 4.8.1 Assume the following overloaded declarations

```
def double (arg: Number): Number
def double (arg: Integer): Integer
```

Now, the following method application is invalid, because two functions resolve to the same arguments set:

```
// variable-less:
double 42
```

Now, with explicitly applied argument types, the following method applications are correct:

```
// variable-less:
double 42 as Integer

// with a variable:
var number: Integer := 42
double number

var number := 42 // the type is inferred
double number
```

4.9 Use Clauses

Syntax:

```
Use             ::= 'use' Use_Expr
Use_Expr        ::= (Container_Path | Stable_Id) '.' Import_Expr
Import_Expr     ::= Single_Import
                  | '{' Import_Exprs '}'
                  | '_'
Import_Exprs    ::= Single_Import {',' Single_Import} [',' '_']
Single_Import   ::= importable_id ['as' [constant_id | '_']]
Container_Path  ::= Module_Path ['.' Constant_Path]
                  | Constant_Path
Module_Path     ::= Module_Selector {'.' Module_Selector}
Constant_Path   ::= Const_Selector {'.' Const_Selector}
Module_Selector ::= constant_id [Vendor_Arg]
Const_Selector  ::= constant_id
Vendor_Arg      ::= '~[' vendor_domain ']'
vendor_domain   ::= vendor_char {vendor_char}
vendor_char     ::= lower | '.' | '-' | '_'
```

A use clause has the form **use** p**.**I, where p is a path to the containing type of the imported entity (either a module or another class), and I is an import expression. The import expression determines a set of names (or just one name) of *importable members*[13] of p, which are made available without full qualification, e.g. as an unqualified name. A member m of p is *importable*, if it is *visible* from the import scope and not object-private (§5.2). The most general form of an import expression is a list of *import selectors*

{ x_1 **as** y_1 ,..., x_n **as** y_n**,** _ }

for $n \geq 0$, where the final wildcard "_" may be absent. It makes available each importable member $p.x_i$ under the unqualified name y_i. I.e. every import selector

[13] Dynamically created members are not importable, since the compiler has no way to predict their existence.

x_i **as** y_i renames (aliases) $p.x_i$ to y_i. If a final wildcard is present, all importable members z of p other than $x_1, \ldots, x_n, y_1, \ldots, y_n$ are also made available under their own unqualified names.

Import selectors work in the same way for type and term members. For instance, a use clause **use** $p.\{x$ **as** $y\}$ renames the term name $p.x$ to the term name y and the type name $p.x$ to the type name y. At least one of these two names must reference an importable member of p.

If the target name in an import selector is a wildcard, the import selector hides access to the source member. For instance, the import selector x_i **as** _ "renames" x to the wildcard symbol, which basically means discarding the name, since _ is not a readable name[14], and thereby effectively prevents unqualified access to x. This is useful if there is a final wildcard in the same import selector list, which imports all members not mentioned in previous import selectors, to selectively not import some members.

The scope of a binding introduced by an import-clause starts immediately after the import clause and extends to the end of the enclosing scope and all nested scopes.

Several shorthands exists. An import selector may be just a simple name x, in which case, x is imported without renaming, so the import selector is equivalent to x **as** y. Furthermore, it is possible to replace the whole import selector list by a single identifier of wildcard. The use clause **use** $p.x$ is equivalent to **use** $p.\{x\}$, i.e. it makes available without qualification the member x of p. The use clause **use** $p._$ is equivalent to **use** $p.\{\_\}$, i.e. it makes available without qualification all importable members of p (this is analogous to **import** $p.*$ in Java or **import** $p._$ in Scala).

Example 4.9.1 Consider the object definition:

```
object M
  def z := 0
  def one := 1
  def add (x: Integer, y: Integer): Integer := x + y
end
```

Then the block

```
{ use M.{one, z as zero, _}; add (zero, one) }
```

is equivalent to the block

```
{ M.add (M.z, M.one) } .
```

[14]Meaning, it is not possible to use "_" as a variable to read from, it never has any value.

Chapter 5

Classes & Objects

Syntax:

```
Tmpl_Def ::= ['case'] 'class' Class_Def
           | ['case'] 'object' Object_Def
           | 'trait' Trait_Def
           | 'module' Module_Def
           | 'protocol' Pro_Def
           | ['case'] 'interface' Ifc_Def
           | 'refinement' Refinement_Def
           | 'type' Const_Type_Def 'end' ['type']
```

Classes (§5.3) & objects (§5.4) are both defined in terms of *templates*.

5.1 Templates

Syntax:

```
Class_Parents  ::= Constr {['prepend'] 'with' Annot_Type}
Trait_Parents  ::= Annot_Type {['prepend'] 'with' Annot_Type}
Template_Body  ::= [Self_Type] Template_Stat {semi Template_Stat}
Self_Type      ::= 'requires' 'self' ':' Type semi
                 | 'requires' [variable_id ':'] Type semi
                 | 'use' ['self'] 'as' variable_id [':' Type] semi
```

A template defines the type signature, behaviour and initial state of a trait, class of objects or of a single object. Templates for part of instance creation expressions (constructors, see §5.1.1 & §5.3.1), class definitions and object definitions. A template *sc* **with** mt_1 **with** ... **with** mt_n { *stats* } consists of constructor invocation *sc*, which defines the template's *superclass*, trait references $mt_1, \ldots, mt_n$ ($n \geq 0$), which

statically define the template's included traits[1], and a statement sequence *stats*, which contains initialization code and additional member definitions & declarations for the template. Unlike in Scala, all trait references in class/trait parents need not to be exhaustive, as more prepended/included traits may be defined as a part of the template body. Trait references declared using **`prepend with`** are prepended to the template body instead of included (§5.1.4).

Each trait reference mt_i that is not prepended must denote a trait (§5.3.4). By contrast, the superclass constructor *sc* normally refers to a class which is not a trait. It is possible to write a list of parents that starts with a trait reference, e.g. mt_1 **`with`** ... **`with`** mt_n . In that case, the list of parents is implicitly extended to include the supertype of mt_1 as first parent type. This new supertype must have at least one constructor that does not take parameters and is accessible to the subclass (§5.2).

The list of parents of a template must be well-formed, i.e. the class denoted by the superclass constructor *sc* must be a subclass (or the superclass itself) of the superclasses of all the traits $mt_1, \ldots, mt_n$.

The *least proper supertype* of a template is the class type or compound type (§3.3.7) consisting of all its parent class types.

The statement sequence *stats* contain member definitions that define new members or overwrite members in the parent classes. It is called also the *class-level block*, as it does not need to contain only member definitions for the template, but also arbitrary other expressions that construct the class object and that are executed while the class is being loaded, in the context of the class. If the template forms part of an abstract class or trait definition, the statement part *stats* may also contain declarations of abstract members. If the template forms part of a concrete class definition, *stats* may still contain declarations of abstract type members, but not of abstract term members. Unlike in Scala, the expressions in *stats* are not forming the primary constructor of the class, but a multi-constructor[2] of the class itself.

The sequence of template statements may be prefixed with a formal parameter definition prefixed with **`requires`** or **`use`**, i.e. **`use self as`** x, **`use self as`** x**`:`** T, **`requires`** T or **`requires`** x**`:`** T. If a formal parameter x is given, it can be used as an alias for the reference **`self`** throughout the body of the template, including any nested types. If the formal parameter x comes with a type T, this definition affects the *self type* S of the underlying class or objects as follows: Let C be the type of the class or trait or object defining the template. If a type T is given for the formal self parameter, S is the greatest lower bound of T and C. If no type T is given, S is simply C. Inside the template, the type of **`self`** is assumed to be S.

The self type of a class or object must conform to the self types of all classes which are inherited by the template t.

[1] Including protocols, which are also traits.

[2] The classes are open in Coral, a single class may have its statements spread across multiple source files.

A second form of self type definition reads just **`requires self:`** *S* . It prescribes the type *S* for **`self`** without introducing an alias name for it.

Example 5.1.1 Consider the following class definitions:

```
class Base extends Object; ... end
trait Mixin extends Base; ... end
object O extends Mixin; ... end
```

In this case, the definition of `O` is expanded to be:

```
object O extends Base with Mixin; ... end
```

5.1.1 Constructor Invocations

Syntax:

```
Constr ::= Annot_Type {'(' [Exprs] ')'}
         | '(' ')'
```

Constructor invocations define the type, members and initial state of objects created by an instance creation expression, or of parts of an object's definition, which are inherited by a class or object definition. A constructor invocation is a function application *c*`[`*targs*`]@[`*dargs*`] [<`*uomargs*`>]`$(args_1)\ldots(args_n)$, where *c* is a path to the superclass or an alias for the superclass, *targs* is a type argument list, *dargs* is an indexing argument list, *uomargs* is a units of measure argument list, $args_1, \ldots, args_n$ are argument lists, and there is a constructor of that class which is applicable to the given arguments.

A type argument list can be only given if the class *c* takes type parameters. An indexing argument list can be only given if the class *c* takes indexing parameters. A units of measure argument list can be only given if the class *c* is a class aggregated with units of measure. If no explicit arguments are given, an empty list `()` is implicitly supplied, unless an explicit primary constructor definition is given, calling explicitly a superconstructor – in that case, the constructor invocation only defines the superclass, and the invocation itself is deferred to the explicit primary constructor.

The superclass constructor is implicitly invoked before any code that the primary constructor defines, but not before early definitions are evaluated.

5.1.2 Metaclasses & Eigenclasses

Metaclasses. A *metaclass* is a class whose instances are classes. Just as an ordinary class defines the behavior and properties of its instances, a metaclass defines the behavior of its class. Classes are first-class citizens in Coral.

Everything is an object in Coral. Every object has a class that defines the structure (i.e. the instance variables) and behavior of that object (i.e. the messages the object can receive and the way it responds to them). Together this implies that a class is an object and therefore a class needs to be an instance of a class (called metaclass).

Class methods actually belong to the metaclass, just as instance methods actually belong to the class. All metaclasses are instances of only one class called `Metaclass`, which is a subclass of the class `Class`.

In Coral, every class (except for the root class `Object`) has a superclass. The base superclass of all metaclasses is the class `Class`, which describes the general nature of classes.

The superclass hierarchy for metaclasses parallels that for classes, except for the class `Object`. The following holds for the class `Object`:

```
Object.class = Class[Object]
Object.superclass = nil
```

Classes and metaclasses are "born together". Every `Metaclass` instance has a method `this_class`, which returns the conjoined class.

Eigenclasses. Coral further purifies the concept of metaclasses by introducing *eigenclasses*, borrowed from Ruby, but keeping the `Metaclass` known from Smalltalk-80. Every metaclass is an eigenclass, either to a class, to a terminal object, or to another eigenclass[3].

Table 5.1: Of objects, classes & eigenclasses

Classes	Eigenclasses of classes	Eigenclasses of eigenclasses
Terminal objects	Eigenclasses of terminal objects	

Eigenclasses are manipulated indirectly through various syntax features of Coral, or directly using the `eigenclass` method. This method can possibly trigger creation of an eigenclass, if the receiver of the `eigenclass` message did not previously have its own (singleton) eigenclass (because it was a terminal object whose eigenclass was a regular class, or the reciever was an eigenclass itself).

Another way to access an eigenclass is to use the **`class`** `<< obj; ...;` **`end`** construct. The block of code inside runs is evaluated in the scope of the eigenclass of `obj`.

[3]Eigenclasses of eigenclasses ("higher-order" eigenclasses) are supposed to be rarely needed, but are there for conceptual integrity, establishing infinite regress.

Metaclass Access. Metaclasses of classes may be accessed using the following language construct.

Syntax:

```
Metaclass_Access ::= 'class' '<<' Metaclass_Obj semi
                     [Exprs] 'end'
Metaclass_Obj    ::= Type | Path | 'self' | variable_id
```

Example 5.1.2 The following code shows how metaclasses are nested in case of `Object` type. Don't try this at home though.

```
class << Object
  self = Metaclass[Object]
  class << self
    self = Metaclass[Metaclass[Object]]
    class << self
      self = Metaclass[Metaclass[Metaclass[Object]]]
    end
  end
end
```

Example 5.1.3 The following code shows what **`self`** references when inside of a class definition, but outside of any defined methods.

```
class Object extends ()
  self = Class[Object]
  class << self
    self = Metaclass[Object]
  end
end
```

Example 5.1.4 Direct access to the eigenclass of any object, here a class' eigenclass:

```
class A
begin
  class << self
    def a_class_method
      "A.a_class_method"
    end def
  end
end class
```

Class `A` uses the **`class`** `<< obj; ...;` **`end`** construct to get direct access to the eigenclass. The keyword **`self`** inside the block is bound to the eigenclass object.

Example 5.1.5 Alternative direct access to the eigenclass of any object, here a class' eigenclass:

```
class B
begin
  self.eigenclass do
    def a_class_method
      "B.a_class_method"
    end def
  end
end class
```

Class `B` uses the `eigenclass` method, which—given a block—evaluates the block in the scope of the eigenclass of **`self`**, which is bound to the class `B`. The keyword **`self`** inside the block is again bound directly to the eigenclass object.

Example 5.1.6 Indirect access to the eigenclass using a singleton method definition:

```
class C
begin
  def self.a_class_method
    "C.a_class_method"
  end def
end class
```

Class `C` uses singleton method definition to add methods to the eigenclass of the class `C`. The keyword **`self`** is bound to the class object in the class-level block and in the new method as well, but the eigenclass is accessed only indirectly.

Example 5.1.7 Indirect access to the eigenclass using a class object definition:

```
class D
begin
  object D
    def a_class_method
      "D.a_class_method"
    end def
  end object
end class
```

Class `D` uses the recommended approach, utilizing standard ways of adding methods to the eigenclass of the class `D`. Here, the eigenclass instance itself is not accessed directly.

Example 5.1.8 Alternative indirect access to the eigenclass using a class object definition:

```
object E
  def a_class_method
    "E.a_class_method"
  end def
end object
```

Class `E` uses a similar recommended approach, utilizing standard ways of adding methods to the eigenclass of the class `E` and neither declaring nor defining anything for its own instances. Here, the eigenclass instance itself is not accessed directly.

5.1.3 Class Linearization

The classes reachable through transitive closure of the direct inheritance relation from a class C are called the *base classes* of C. Because of traits, the inheritance relationship on base classes forms in general a directed acyclic graph. A linearization of this graph is defined as follows.

Definition 5.1.9 Let base classes of a class C be the list of every superclass of C with every trait that these classes include and/or prepend and every protocol that these classes implement. Let C be a class with base classes C_1 **with** C_2 **with** … **with** C_n. The *linearization* of C, $\mathscr{L}(C)$ is defined as follows:

$$\mathscr{L}(C) \quad = \quad C\ ,\ \mathscr{L}(C_n) \vec{+} \ldots \vec{+} \mathscr{L}(C_1)$$

Here $\vec{+}$ denotes concatenation, where elements of the right operand replace identical elements of the left operand:

$$\begin{array}{rcll} \{a, A\} \vec{+} B & = & a, (A \vec{+} B) & \textbf{if } a \notin B \\ & = & A \vec{+} B & \textbf{if } a \in B \end{array}$$

Example 5.1.10 Consider the following class definitions.[4]

```
class Abstract_Iterator extends Object; ... end
trait Rich_Iterator extends Abstract_Iterator; ... end
class String_Iterator extends Abstract_Iterator; ... end
class Iterator extends String_Iterator with Rich_Iterator; ... end
```

Then the linearization of class `Iterator` is

```
{ Iterator, Rich_Iterator, String_Iterator, Abstract_Iterator,
  Object }
```

Note that the linearization of a class refines the inheritance relation: if C is a subclass of D, then C precedes D in any linearization where both C and D occur. Also note that whether a trait is included or prepended is irrelevant to linearization, but essential to function applications (§6.6).

[4]Here we say "class", but that term includes now traits as well.

5.1.4 Inheritance Trees & Include Classes

Include classes. A mechanism that allows arbitrary including and prepending of trait into classes and inheritance binary trees[5] uses a transparent structure called *include class*. Include classes are always defined indirectly.

Every class has a link to its superclass. In fact, the link is made up of an include class structure, which itself holds an actual link to the superclass. That superclass has its own link to its superclass and this chain goes forever until the `Object` class is encountered, which has no superclass.

Included traits. When a trait M is included into a class, a new include class Im_i is inserted between the target class and the include class Is that holds a link to its superclass (or to a previously included trait Im_{i+1}). This include class Im_i then holds a link to the included trait. Every class that includes the trait M is available via the `included_in` method of M.

If a trait M is already (included in or prepended to)[6] any of the superclasses, then it is not included again. Included traits act like superclasses of the class they are included in, and they *overlay* the superclasses.

Example 5.1.11 A sample trait schema:

```
class C extends D with Some_Trait
end

C → [Some_Trait] → [D]
D → [Object]
```

Include classes are depicted by the brackets, with their link value inside. Note that include classes only know their super-type (depicted by the arrow "→") and their link value (inside brackets).

Prepended traits. When a trait M is prepended to a class, a new include class Im_i is inserted between the target class and its last prepended trait, if any. Prepended traits are stored in a secondary inheritance chain just for prepended traits, forming an inheritance tree. Every class that has M prepended is available via the `prepended_in` method of M. The effect of prepending a trait in a class or a trait is named *overlaying*.

If a trait M is already prepended to any of the superclasses, it has to be prepended again, since the already prepended traits of superclasses are in super-position to the class

[5] Yes, trees, not chains: prepended traits make the inheritance game stronger by forking the inheritance chain at each class with prepended traits, forming a shape similar to a rake.

[6] This is important since both included and prepended traits act like superclasses to every subclass.

or trait that gets M prepended. Prepended traits of superclasses do not *overlay* child classes. Prepended traits are inserted into the inheritance tree more like subclasses than superclasses.

Nested includes. When a trait M itself includes one or more traits $M_1, \ldots, M_n$, then these included traits are inserted between the included trait M and the superclass, unless they are already respectively included by any of the superclasses. If M with included $M_1, \ldots, M_n$ is prepended to C, then $M_1, \ldots, M_n$ are inserted before the superclass of C, if not already included in any of the superclasses. It is an error if nested includes form a dependency cycle: any *auto-included* trait must not require to include a trait that triggered the auto-include. The order in which traits are included in another trait may change when included in a class, i.e. if the class includes two traits A and B that themselves include the same two traits D and E in reverse order (A includes D, then E, but B includes E, then D): the order is then defined by the first trait that included the two auto-included traits and subsequently included traits can not change this order in any way.

Example 5.1.12 Take the following trait and class definitions, where S is the superclass of the class C:

```
trait D end
trait E end

trait A extends D with E; end
trait B extends E with D; end

class C extends S with A with B; end
```

Then traits are auto-included in the following order:

- `C → [S]`
  First, the superclass is added.
- `C → [A] → [S]`
  Then, trait A is included, unless already included in S.
- `C → [A] → [E] → [S]`
  Including of trait A triggers auto-include of traits included in A. Start with the first one in chain of A: E.
- `C → [A] → [E] → [D] → [S]`
  Then, A has D in its chain.
- `C → [B] → [A] → [E] → [D] → [S]`
  Finally include B. B triggers auto-include, but both of its included traits are already included in the chain, so nothing more happens.

Note that the order of E and D is reversed, since later includes move the trait closer to the including class or trait, and therefore **`super`** calls go through traits that were included before. Also, if B was included sooner than A, then D and E would appear in reverse order in the chain.

Nested prepends. When a trait M itself prepends one or more traits $M_1, \ldots, M_n$, then nothing happens to the class or the trait that M is included in. If M is prepended to C, then every trait $M_1, \ldots, M_n$ prepended to M is automatically prepended to C in this way:

1. Establish a list of traits that triggered auto-prepend, named here tp. This list is in ideal case empty, so it's actually ok for the runtime to wait with its creation until needed and only increase its size in very small steps.
2. Establish a list of traits that are scheduled to be auto-prepended, named here sp. This list is in ideal case empty, again.
3. There is a *prepend chain* in the class that prepends M. If no trait was prepended so far, create it. The chain's head is the element that is the farthest from the class C, the chain's tail is right before the class C. Traits closer to this chain's head are searched for method overlays sooner in runtime than traits closer to the tail (and the class respectively).
4. Insert M at the chain's head, unless M already is in the chain. If it is, then halt.
5. If M has itself prepended traits, insert M into tp (triggered prepend).
6. For traits $M_1, \ldots, M_n$ that are prepended in M, test if each M_i already appears in the chain. If it does, move M up the chain towards the chain's tail, right until M is closer to the chain's tail than M_i. If it does not, add M_i to sp, unless M_i is in tp. If it is, then it is an error[7].
7. If M has included traits, include them in C in the already described way now.
8. If sp is not empty, then for each M_i in sp, remove M_i from sp and recursively apply steps starting with 4 on it. Keep both sp and tp shared for recursive calls. If M_i moves closer to the chain's tail than M or any other trait prepended in prepending of the original M[8], it is an error[9].

Traits & Metaclasses. Since traits may contain "class" methods as well as instance methods, all the operations with include classes are mirrored on the respective metaclasses.[10]

[7] Trait cycle dependency detected.

[8] This can be achieved by having a third list of traits that were prepended.

[9] Trait composition design flaw detected.

[10] This makes Coral in no need of constructs like **`module`** `ClassMethods`, known from Ruby.

5.1.5 Class Members

A class C defined by class definition can define members in its class-level block, can inherit members from all parent classes[11] and included traits, and can have its members overlayed by all prepended traits.

Member definitions fall basically into two categories: *concrete* & *abstract*. Members of the class C are either *directly defined*, *overlayed* or *inherited*.

Concrete members are those that have a definition, abstract members are those that are only declared without any subsequent or preceding definition.[12]

Members are instance variables, class instance variables, types, methods & messages.

5.1.6 Overriding

A member M of a class C that matches a member M' of a base class of C is said to *override* that member. In this case the binding of the overriding member M must conform (§3.5.2) to the binding of the overridden member M'. Furthermore, the following restrictions on modifiers apply to M and M':

- M' must not be labeled **`final`**.
- M must not be **`private`**.
- If M is labeled **`private`**`[`C`]` for some enclosing class or module C, then M' must be labeled **`private`**`[`C'`]`, where C' equals C or C is contained in C'.
- If M is labeled **`protected`**, then M' must also be labeled **`protected`**.
- If M' is labeled **`protected`**, then M' must also be labeled **`protected`** or **`public`**.
- If M' is not an abstract member, then M should be labeled **`override`** or annotated `@{Override}`. Furthermore, one of the possibilities must hold:
  - either M is defined in a subclass of the class where M' is defined,
  - or both M and M' override a third member M'', which is defined in a base class of both the classes containing M and M'.
- If M' is labeled **`private`**, then M' must be labeled **`private`**, **`protected`** or **`public`**.
- If M' is incomplete (§5.2) in C, then M should be labeled **`abstract override`**.
- If M and M' are both concrete value definitions, then either none of them is marked **`lazy`**, or both must be marked **`lazy`**.

[11] Including an interface and any number of protocols.

[12] There is a special case, when a protocol introduces an abstract member, but the class that implements this protocol already inherited or directly defined the matching member.

To generalize the conditions, the modifier of M must be the same or less restrictive than the modifier of M'.

An overriding method, unlike in Scala, does not inherit any default arguments from the definition in the superclass, but, as a convenience, if the default argument is specified as '_', then it gets inherited.

5.1.7 Inheritance Closure

Let C be a class type. The *inheritance closure* of C is the smallest set $\mathscr{S}$ of types such that

- If T is in $\mathscr{S}$, then every type T' which forms syntactically a part of T is also in $\mathscr{S}$.
- If T is a class type in $\mathscr{S}$, then all parents of T (§5.1) are also in $\mathscr{S}$.

It is an error if the inheritance closure of a class type consists of an infinite number of types.

5.1.8 Early Definitions

Syntax:

```
Early_Defs ::= '{' [Early_Def {semi Early_Def}] '}'
               'with'
Early_Def  ::= {Annotation} {Modifier} Pat_Var_Def
```

A template may start with an *early definition* clause, which serves to define certain field values before the supertype constructor is called. In a template

```
{
  val p1: T1 := e1
  ...
  val pn: Tn := en
} with sc with mt1 with ... with mtn
```

The initial pattern definitions of $p_1, \ldots, p_n$ are called *early definitions*. They define fields which form part of the template. Every early definition must define at least one field.

Any reference to **self** in the right-hand side of an early definition refers to the identity of **self** just outside the template, not inside of it. As a consequence, it is impossible for any early definition to refer to the object being constructed by the template, or refer to any of its fields, except for any other preceding early definition in the same section.

5.2 Modifiers

Syntax:

```
Modifier         ::= Local_Modifier
                   | Access_Modifier
                   | 'override'
Local_Modifier   ::= 'implicit'
                   | 'lazy'
                   | 'final'
                   | 'sealed'
                   | 'abstract'
Access_Modifier  ::= 'public'
                   | ('protected' | 'private') [Access_Qualifier]
Access_Qualifier ::= '[' (constant_id | 'self') ']'
```

Access modifiers may appear in two forms:

Accessibility flag modifier. Such a modifier appears in a template (§5.1) or a module definition (§5.5) alone on a single line. All subsequent members in the same class-level block than have accessibility of this modifier applied to them, if allowed to (does not apply to destructors). Only access modifiers can be used this way.

Directly applied modifier. Such a modifier appears on a line preceding a member to which the modifier is solely applied, or a list of arguments with symbols that the modifier will be applied to. A directly applied modifier expression has a return type of `Symbol`, so that it may be used as a regular function and chained.

Example 5.2.1 An example of an accessibility flag modifier:

```
class C
begin
  public
    def hello; end
  private
    def private_hello; end
end class
```

Example 5.2.2 An example of directly applied modifier:

```
class C
begin
  def hello; end
```

```
    def private_hello; end
    def salute; end
    public :hello
    private :private_hello, :salute
    protected def goodbye; end
  end class
```

By default[13], the **public** access modifier affects every member of the class type, except for instance variables and class instance variables, which are object-private (**private**[**self**]).

Modifiers affect the accessibility and usage of the identifiers bound by them. If several modifiers are given, their order does not matter, but the same modifier may not occur more than once and combinations of **public**, **protected** & **private** are not allowed (using them as accessibility flag modifiers overwrites the previous accessibility, not combines them). If a member declaration has a modifier applied to it, then the subsequent member definition has the same modifier already applied to it as well, without the need to explicitly state that. It is an error if the modifier applied to the member definition would contradict the modifier applied to the member declaration.

Accessibility modifiers can not be applied to instance variables and class instance variables (both declarations and definitions). These are by default *instance-private*. This is a sort of relaxation in access restriction, say, every method that is at least *public* and at most *object-private* restricted, and that has the instance as a receiver, can access the instance variable or the class instance variable. Any other method that does not have the particular instance as the receiver, does not have any access to the instance variable or the class instance variable, even if the method is a method of the same class as the particular instance.

The rules governing the validity and meaning of a modifier are as follows:

- The **private** access modifier can be used with any declaration or definition in a class. Such members can be accessed only from within the directly enclosing class, the class object (§5.4) and any member of the directly enclosing class, including inner classes. They are not inherited by subclasses and they may not override definitions in parent classes.

  The modifier may be *qualified* with an identifier C (e.g. **private**[C]) that must denote a class or a module enclosing the declaration or definition. Members labeled with such a modifier are accessible respectively only from code inside the module C or only from code inside the class C and the class object C (§5.4).

  A different form of qualification is **private**[**self**]. A member M marked with this modifier is called *object-private*; it can be accessed only from within the object in which it is defined. That is, a selection $p.M$ is only legal if the prefix ends

[13]That is, without any explicit modifier being applied.

with **this** or **self** and starts with O for some class O enclosing the reference. . Moreover, the restrictions for unqualified **private** apply as well.

Members marked **private** without any qualifier are called *class-private*. A member *is private* if it is either class-private or object-private, but not if it is marked **private**[C], where C is an identifier, in the latter case the member is called *qualified private*.

Class-private and object-private members must not be **abstract**, since there is no way to provide a concrete implementation for them, as private members are not inherited. Moreover, modifiers **protected** & **public** can not be applied to them (that would be a contradiction[14]), and the modifier **override** can not be applied to them as well[15].

- The **protected** access modifier can be used with any declaration or definition in a class. Protected members of a class can be accessed from within:
  - the defining class
  - all classes that have the defining class as a base class
  - all class objects of any of those classes

  A **protected** access modifier can be qualified with an identifier C (e.g. **protected**[C]) that must denote a class or module enclosing the definition. Members labeled with such a modifier are *also*[16] accessible respectively from all code inside the module C or from all code inside the class C and its class object C (§5.4).

  A protected identifier x can be used as a member name in a selection r**.**x only if one of the following applies:

  - The access is within the class defining the member, or, if a qualification C is given, inside the module C, the class C or the class object C, or
  - r ends with one of the keywords **this**, **self** or **super**, or
  - r's type conforms to a type-instance of the class which has the access to x.

  A different form of qualification is **protected**[**self**]. A member M marked with this modifier can be accessed only from within the object in which it is defined, including methods from inherited scope. That is, a selection p**.**M is only legal if the prefix ends with **this**, **self** or **super** and starts with O for some class O enclosing the reference. Moreover, the restrictions for unqualified **protected** apply.

[14] E.g., a member can not be public and private at the same time.

[15] Otherwise, if a private member could override an inherited member, that would mean there is an inherited member that could be overridden, but private members can not override anything: only protected and public members can be overridden. If a member was overriding an inherited member, the parent class would *lose access* to it.

[16] In addition to unqualified **protected** access.

- The **`override`** modifier applies to class member definitions and declarations. It is never mandatory, unlike in Scala or C# (in further contrast with C#, every method in Coral is virtual, so Coral has no need for a keyword "`virtual`"). On the other hand, when the modifier is used, it is mandatory for the superclass to define or declare at least one matching member (either concrete or abstract).

- The **`override`** modifier has an additional significance when combined with the **`abstract`** local modifier. That modifier combination is only allowed for members of traits.

  We call a member *M* of a class or trait *incomplete* if it is either abstract (i.e. defined by a declaration), or it is labeled **`abstract`** and **`override`** and every member overridden by *M* is again incomplete.

  The **`abstract override`** modifier combination does not influence the concept whether a member is concrete or abstract. A member is *abstract* if only a declaration is given for it; it is *concrete* if a full definition is given. This behavior can be turned off only in tests, if needed, and is implicitly turned on.

  The **`abstract override`** modifier combination can be thus used with a full definition in a trait and yet affect the class or trait with which it is used, so that a member access to member **`abstract override`***M*, such as **`super.`***M*, is legal. But, the **`abstract override`** modifier combination does not need to be applied to a definition, a declaration is good enough for it.

  Additionally, an annotation `@[Override]` exists for class members that triggers only warnings in case the member has no inherited member to override, but does not prevent the class from being created. Thus, the annotation only signals an intention, the keyword modifier signals a requirement.

- The **`abstract`** local modifier is used in class declarations. It is never mandatory for classes with incomplete members or for declarations and definitions. It is implied (and therefore redundant) for traits. Abstract classes can not be instantiated (an exception is raised if tried to do so), unless provided with traits and/or a refinement which override all incomplete members of the class. Only abstract classes and (all) traits can have abstract term members. This behaviour can be turned off only in tests, if needed, and is implicitly turned on.

  The **`abstract`** local modifier can be used with conjunction with **`override`** modifier for class member definitions.

  Additionally, an annotation `@[Abstract]` exists for classes and class members that triggers only warnings in case of instantiation, but does not prevent the instantiation. Thus, the annotation only signals an intention, the keyword modifier signals a requirement.

- The **`final`** local modifier applies to class members definitions and to class definitions. Every **`final`** class member can not be overridden in subclasses. Every **`final`** class can not be inherited by a class or trait. Members of final classes are

implicitly also final. Note that **`final`** may not be applied to incomplete members, and can not be combined in one modifier list with the **`sealed`** local modifier.

Additionally, an annotation `@[Final]` exists for classes and class members that triggers only warnings in case of inheriting or overriding respectively, but does not prevent the inheritance or overriding respectively. Thus, the annotation only signals an intention, the keyword modifier signals a requirement.

- The **`sealed`** local modifier applies to class definitions. A **`sealed`** class can not be directly inherited, except if the inheriting class or trait is defined in the same source file as the inherited sealed class. However, subclasses of a sealed class have no restriction in inheritance, unless they are final or sealed again.

  Additionally, an annotation `@[Sealed]` exists for classes and class members that triggers only warnings in case of inheriting outside the same source file, but does not prevent the inheritance. Thus, the annotation only signals an intention, the keyword modifier signals a requirement.

- The **`lazy`** local modifier applies to value definitions. A **`lazy`** value is initialized the first time it is accessed (which might eventually never happen). Attempting to access a lazy value during its initialization is a blocking invocation until the value is initialized of failed to initialize. If an exception is thrown during initialization, the value is considered uninitialized and the initialization is restarted on later access, re-evaluating its right hand side.

Example 5.2.3 The following code illustrates the use of qualified and unqualified private:

```
module Outer_Mod.Inner_Mod
  class Outer
  begin
    class Inner
    begin
      private[self] def e() end def
      private def f() end def
      private[Outer] def g() end def
      private[Inner_Mod] def h() end def
      private[Outer_Mod] def i() end def
    end class
  end class
end module
```

Here, accesses to the method `e` can appear anywhere within the instance of `Inner`, provided that the instance is also the receiver at the same time. Accesses to the method `f` can appear anywhere within the class `Inner`, including all receivers of the same class. Accesses to the method `g` can appear anywhere within the

class `Outer`, but not outside of it. Accesses to the method `h` can appear anywhere within the module `Outer_Mod.Inner_Mod`, but not outside of it, similar to package-private methods in Java. Finally, accesses to the method `h` can appear anywhere within the module `Outer_Mod`, including modules and classes contained in it, but not outside of these.

A rule for access modifiers in scope of overriding: Any overriding member may be defined with the same access modifier, or with a less restrictive access modifier. No overriding member can have more restrictive access modifier, since the parent class would *lose access* to the member, and that is unacceptable.

- Modifier **`public`** is less restrictive than any other access modifier.
- Qualified modifier **`protected`** is less restrictive than an unqualified **`protected`**, only if the class that the modifier is qualified with is among base classes of the original class – the original class must not lose access.
- Qualified modifier **`protected`** is less restrictive than object-protected, only if the class that the modifier is qualified with is among base classes of the original class – the original class must not lose access.
- While **`protected`** is certainly less restrictive than **`private`**, private members are not inherited and thus can not be overridden.
- While qualified **`private`** is certainly less restrictive than unqualified **`private`**, private members are not inherited and thus can not be overridden.

The relaxations of access modifiers for overriding members are then available as follows:

- **`protected[self]`** → { **`protected`**, **`protected[`**C**`]`**, **`public`** }
- **`protected`** → { **`protected[`**C**`]`**, **`public`** }
- **`protected[`**C**`]`** → { **`protected[`**D**`]`**, **`public`** }

  This is only for the case where C is accessible from within D.
- **`protected[`**C**`]`** → { **`public`** }
- **`public`** → { **`public`** }

  This is just for the sake of completeness, since change from public to public is not much of a relaxation.

5.3 Class Definitions

Syntax:

```
Tmpl_Def            ::= 'class' Class_Def
Class_Def           ::= constant_id [Type_Param_Clause]
                        [Dep_Params] [UoM_Params]
                        [Class_Param_Clauses] Class_Tmpl_Env
Class_Param_Clauses ::= {Annotation} [Access_Modifier]
                        {Class_Param_Clause}
                        ['(' 'implicit' Class_Params ')']
Class_Param_Clause  ::= '(' [Class_Params] ')'
Class_Params        ::= Class_Param {',' Class_Param}
Class_Param         ::= {Annotation} [{Modifier} ('val' | 'var')]
                        variable_id [':' Param_Type] [':=' Expr]
Class_Tmpl_Env      ::= 'extends' [Early_Defs] Class_Parents
                        semi [Template_Body] 'end' ['class']
                      | ['extends' [Early_Defs]]
                        '{' [Template_Body] '}'
                      | ['begin' [Template_Body]] 'end' ['class']
```

A class definition defines the type signature, behavior and initial state of a class of objects (the instances of the defined class) and of the class object, which is the class instance itself, with behavior defined in its metaclass (§5.1.2).

The most general form of a class definition is

class *c* [*tps*] @[*dps*] [<*uomps*>]
  as $m(ps_1)\ldots(ps_n)$
  extends *t*

for $n \geq 0$.

Here,

c is the name of the class to be defined.

tps is a non-empty list of type parameters of the class being defined. The scope of a type parameter is the whole class definition including the type parameter section itself. It is an error to define two type parameters with the same name. The type parameter section [*tps*] may be omitted. A class with a type parameter section is called *polymorphic*, a class without a type parameter section is called otherwise *monomorphic*. Type arguments are reified in Coral, i.e. type arguments are preserved in runtime[17], generating a new concrete subtype of the original

[17]Unlike in Java or Scala, which both perform type erasure.

generic class. This ensures type safety in a dynamic environment of Coral in runtime.

dps is a non-empty list of indexing parameters for dependent types (§5.8). The indexing parameter section `@[`*dps*`]` may be omitted. A class with an indexing parameter section is called *indexed*. Indexing arguments are reified in Coral, i.e. indexing arguments are preserved in runtime.

uomps is a non-empty list of unit of measure parameters for aggregated types (§5.9). The unit of measure parameter section `[<`*uomps*`>]` may be omitted. A class with a unit of measure parameter section is called *aggregated with units of measure*. Units of measure arguments are reified in Coral, i.e. units of measure arguments are preserved in runtime.

as is a possibly empty sequence of annotations (§10). If any annotations are given at this point, they apply to the primary constructor of the class.

m is an access modifier (§5.2), such as **`private`** or **`protected`** (**`public`** is implied otherwise), possibly with a qualification. If such an access modifier is given, it applies to the primary constructor of the class.

$(ps_1)\ldots(ps_n)$ are formal value parameter clauses for the *primary constructor* of the class. The scope of a formal value parameter includes all subsequent parameter sections and the template *t*. However, a formal value parameter may not form part of the types of any of the parent classes or members of the class template *t* – only parameters from *tps* may form part of the types of any of the parent classes or members of the class template *t*. It is illegal to define two formal value parameters with the same name. If no formal parameter sections are given, an empty parameter section `()` is implied.

If a formal parameter declaration *x*`:` *T* is preceded by a **`val`** or **`var`** keyword, an accessor (getter) definition (§4.1) for this parameter is implicitly added to the class. The getter introduces a value member *x* of class *c* that is defined as an alias of the parameter. Moreover, if the introducing keyword is **`var`**, a setter accessor *x*`_=(`*e*`)` is also implicitly added to the class. The formal parameter declaration may contain modifiers, which then carry over to the accessor definition(s). A formal parameter prefixed by **`val`** or **`var`** may not at the same time be a call-by-name parameter (§4.7.1).

If a formal value parameter is a part of the definition of a property (a property getter and/or a property value), then these accessors are not implicitly added, as the arguments may be traced into the properties. The same applies to instance value definitions that have a formal value parameter forming a part of it. In any case, every formal value parameter is implicitly added as an instance value definition of a **`val`** or **`var`** type respectively.

t is a template (§5.1) of the form

sc **with** mt_1 **with** ... **with** mt_n { *stats* }

for ($n \geq 0$), which defines the base classes, behavior and initial state of objects of the class. The extends clause **extends** *sc* **with** mt_1 **with** ... **with** mt_n can be omitted, in which case **extends** `Object` is implied. The only class that defines *sc* as `()` is `Object` (meaning that it has no superclass) and it is an error if any other class attempts to do the same.

This class definition defines a type *c*[*tps*]@[*dps*] [<*uomps*>] and a primary constructor, which, when applied to arguments conforming to types *ps*, initializes instances of type *c*[*tps*]@[*dps*] [<*uomps*>] by evaluating the primary constructor parts defined in the template *t*.

Example 5.3.1 The following example illustrates **val** and **var** parameters of a class `C`:

```
class C (x: Integer, val y: String, var z: List[String]) end
val c := C.new(1, "abc", List[String].new)
c.z := c.y ~> c.z
```

Example 5.3.2 The following class can be created only from its class object.

```
class Sensitive private () {
  ...
}
object Sensitive {
  def make_sensitive (credentials: Certificate): Sensitive := {
    if credentials.admin?
      Sensitive.new
    else
      raise SecurityViolationException
    end
  }
}
```

5.3.1 Constructor & Destructor Definitions

Primary constructor. The primary constructor is a special function, more special in its syntax than any other function in Coral. Its definition can spread across three syntactically different places inside of a class definition: formal value parameters, explicit field declarations, explicit primary constructor body. Primary constructor without explicitly defined parameters is equivalent to the default constructor.

Primary constructor parameters. These are described as a part of class definitions (§5.3). Technically, a primary constructor is happy being left with nothing but the parameter definitions, since these are automatically mapped to instance values[18] and also implicitly adds accessor methods to the class with appropriate visibility, defined explicitly by accessor modifiers (§5.2) or implicitly as **`public`**.

Explicit primary constructor fields. These are instance value definitions inside the class template (§5.1) that use the formal value parameters either with or without any modification. Formal value parameter used without any modification is equivalent to the expression defining the value of an instance value being only the parameter itself. If the parameter is used without modification, it becomes an alias to the instance value implicitly defined by the primary constructor parameter. As the explicit field can be expected to be accessed more often than the original parameter, a Coral VM implementation may opt in to make the implicit instance value being the alias, instead of the explicit field.

Example 5.3.3 An example of an explicit primary constructor field.

```
class C (val y: String) {
  val @x := y               // without modification
  val @z := y + "_modified"  // with modification
}
```

Explicit primary constructor body. If a primary constructor should do something more than simply map parameters to instance values, then its explicit body comes in. It may co-exist with explicit primary constructor fields, which are then executed right after a call to the super-constructor (either in implicit position, or an explicit one). **Syntax:**

```
Def             ::= PCtor_Def
PCtor_Def       ::= 'constructor' PCtor_Fun_Def
PCtor_Fun_Def   ::= '(' '_' ')' (':=' Constr_Expr | Constr_Block)
Constr_Expr     ::= Self_Invocation
                  | Constr_Block2
Self_Invocation ::= 'self' Argument_Exprs {Argument_Exprs}
                  | Sup_Invocation
Sup_Invocation  ::= 'super' [Argument_Exprs {Argument_Exprs}]
Constr_Block    ::= Constr_Block1 | Constr_Block2
Constr_Block1   ::= [semi Constr_Block3] 'end' ['constructor']
Constr_Block2   ::= '{' [Constr_Block3] '}'
Constr_Block3   ::= [[Block_Stat {semi Block_Stat} semi]
                     Self_Invocation semi]
                     Block_Stat {semi Block_Stat}
                  | Self_Invocation
```

[18] Making this behavior overrideable may be discussed, but it might interfere with case classes.

Auxiliary constructors. Besides the primary constructor, a class may define additional constructors with different parameter sections, that form together with the primary constructor an overloaded constructor definition. Every auxiliary constructor is constrained in means that it has to either invoke another constructor defined before itself, or any superclass constructor. Therefore classes in Coral have multiple entry point, but with great power comes great responsibility: user must ensure that every constructor path defines all necessary members. It is an error if a constructor invocation leads to an instance with abstract members. If a constructor does not explicitly invoke another constructor, an invocation of the implicit constructor is inserted implicitly as the first invocation in the constructor.

Syntax:

```
Ctor_Def     ::= 'constructor' Ctor_Fun_Def
Ctor_Fun_Def ::= [Param_Clauses] (':=' Constr_Expr | Constr_Block)

Dtor_Def     ::= 'destructor' Dtor_Fun_Def
Dtor_Fun_Def ::= (':=' Dtor_Expr | Dtor_Block)
Dtor_Expr    ::= Sup_Invocation
               | Dtor_Block2
Dtor_Block   ::= Dtor_Block1 | Dtor_Block2
Dtor_Block1  ::= [semi Dtor_Block3] 'end' ['destructor']
Dtor_Block2  ::= '{' [Dtor_Block3] '}'
Dtor_Block3  ::= [[Block_Stat {semi Block_Stat} semi]
                   Sup_Invocation semi]
                   Block_Stat {semi Block_Stat}
                 | Sup_Invocation
```

Default constructor. A default constructor of a class is the explicit parameterless constructor. This differs from Java or C#. The default root constructor of `Object` carries the modifier **`public`** (§5.2), and therefore to explicitly disallow direct object creating, custom constructors have to be explicitly more restrictive than **`public`**.

Implicit constructor. An implicit constructor is an automatically generated bridge constructor to the parameterless default constructor. This is what Java and C# call "default constructor". An implicit constructor "does nothing" but invokes the superconstructor and initializes all members specific to the constructed object to their default values (either implicit one, which is **`nil`**, or explicit ones used in their definitions).

Convenience constructor. A convenience constructor is any other constructor than the parameterless default constructor, including the primary constructor, if it has parameters.

Accessibility of constructors. Constructors may have modified accessibility, so that only certain functions can invoke them indirectly. The accessibility is then transitioned from the calling context.

Example 5.3.4 An example of a convenience constructor of class C.

```
class C
  constructor (param)
    // super is invoked implicitly here
    val @resource := param
  end constructor
end class
```

Example 5.3.5 An example of a pair of constructors of class C.

```
class C
  constructor := self(42)
  constructor (param)
    // super is invoked implicitly here
    val @resource := param
  end constructor
end class
```

Explicit destructor. An explicit destructor does not have any accessibility. The super-destructor is invoked implicitly at the end of its execution, unless explicitly invoked earlier. Destructors are parameterless and have a further requirement that they can not increment the reference count of the object being destructed – doing so could result in zombie objects.

Implicit destructor. An implicit destructor is an automatically generated bridge destructor to the parameterless super-destructor. An implicit destructor "does nothing" but release all members specific to the destructed object and invoke super-destructor afterwards. The destructor of `Object` releases every remaining member of the destructed object. A class can only have a single destructor, either an explicit or an implicit one.

Accessibility of destructors. Destructors, unlike constructors, can not have any accessibility modifiers. They ignore the current accessibility flag of their class-block and trigger a warning if a modifier is used directly with the destructor. Destructors may be invoked independently on the context in which the object is destructed.

Example 5.3.6 An example of an explicit destructor of class C.

```
class C
  destructor
    @resource.close unless @resource.closed?
    // super is invoked implicitly here
  end destructor
end class
```

5.3.2 Clone Constructor Definitions

Syntax:

```
CCtor_Def     ::= 'clone' CCtor_Fun_Def
CCtor_Fun_Def ::= [Param_Clauses]
                  (':=' Constr_Expr | Constr_Block)
```

Clone constructors are pretty much like constructor, except for they are not invoked indirectly by `allocate` on `Class`, but by **`clone`** on the cloned instance. Regular constructors are not invoked on the cloned objects, since they were already invoked on the original object.

Clone constructor implicitly returns the new cloned object, unless returning explicitly a different object (which would possibly invoke its regular constructors). The original object is available with the **`self`** and **`this`** keywords, the new cloned object is available with the **`self`**`[`**`cloned`**`]` construct. The **`self`**`[`**`cloned`**`]` construct is only legal in a body of the clone constructor. Clone constructors are invoked in the context of the original object, and **`self`**`[`**`cloned`**`]` is the only allowed prefix to instance values and variables, allowing to define new instance values or variables in the clone.

Clone constructors pass on the eigenclass (if any) of the original object to the cloned object, thus elevating it to an almost regular class – a prototype class, a class that resides not in a constant, but in a class instance, in an object (but that original object may be still assigned to a constant anyway).

A different clone constructor of the same class may be invoked by using the **`self`** keyword as a function name. If a clone constructor invokes a different clone constructor of the same class this way, the super-clone-constructor is not implicitly invoked (since it is invoked in the other clone constructors).

Default clone constructor. A default clone constructor of a class is the explicit parameterless clone constructor. The `Object` class has a default root clone constructor, which carries the modifier **`protected`** (§5.2). Therefore, any custom clone constructor has to be explicitly protected in a less restrictive way, or public, in order to allow direct object cloning from outside of the class that defines it.

Implicit clone constructor. An implicit clone constructor is an automatically generated bridge clone constructor to the parameterless default clone constructor. An implicit clone constructor "does nothing" but invokes the super-clone-constructor and makes a shallow copy of every member specific to the cloned object.

Convenience clone constructor. A convenience clone constructor is any other clone constructor than the parameterless default clone constructor. The **clone** method of objects can accept any number of arguments that are then passed into the clone constructor and the clone constructor is resolved based on these passed arguments.

Accessibility of clone constructors. Clone constructors may have modified accessibility, so that only certain functions can invoke them indirectly. The accessibility is then transitioned from the calling context.

Example 5.3.7 An example of a default clone constructor of class C, performing a deep copy.

```
class C
  clone
    // super is invoked implicitly here
    self[cloned].@resource := @resource.clone
  end constructor
end class
```

5.3.3 Case Classes

Syntax:

```
Tmpl_Def ::= 'case' 'class' Class_Def
```

If a class definition is prefixed with **case**, the class is said to be a *case class*.

The formal parameters are handled differently from regular classes. The formal parameters in the first parameter section of a case class are called *elements* and are treated specially:

> First, the value of such a parameter can be extracted as a field of a constructor pattern.
>
> Second, a **val** prefix is implicitly added to such a parameter, unless the parameter already carries a **val** or **var** modifier. Hence, an accessor definition for the parameter is generated (§5.3).

A case class definition of *c*`[`*tps*`]@[`*dps*`] [<`*uomps*`>] (`ps_1`)...(`ps_n`)` with type parameters *tps*, indexing parameters *dps*, aggregated with units of measure *uomps* and value parameters *ps* implicitly adds some methods to the corresponding class object, making it suitable as an extractor object (§8.1.8), and are defined as follows:[19]

```
object c {
  def apply[tps][<uomps>](ps_1)...(ps_n): c[tps] :=
      c[Ts][<UoMs>].new(xs_1)...(xs_n)
  def unapply[tps][<uomps>](x: c[tps][<uomps>]): Option[e] :=
      unless x = nil
        Some(x.xs_11, ..., x.xs_1k)
      else
        None
      end
}
```

Here, *Ts* stands for the vector of types defined in the type parameter section *tps*, *UoMs* stands for the vector of units of measure defined in the units of measure parameter section *uomps*, each xs_i denotes the parameter names of the parameter section ps_i, $xs_{11}, \ldots, xs_{1k}$ denote the names of all parameters in the first parameter section xs_1, and *e* denotes the type that `Option` is parameterized with. If a type parameter section is missing in the class *c*, it is also missing in the `apply` and `unapply` methods. The definition of `apply` is missing also if the class *c* is marked **`abstract`**.[20]

If the case class definition contains an empty value parameter list, the `unapply` returns a `Boolean` instead of an `Option` type and is defined as follows:

```
def unapply[tps][<uomps>](x: c[tps][<uomps>]): Boolean := x != nil
```

For each case of a value parameter list of n parameters, the `unapply` returns these types:

- For $n = 0$, `Boolean` is the return type.
- For $n = 1$, `Option[`Tp_1`]` is the return type, where Tp_1 is the type of the only parameter in the first parameter section.
- For $n \geq 1$, `Option[(`$Tp_1, \ldots, Tp_n$`)]` is the return type, where Tp_i is the type of the i^{th} parameter of the first parameter section. Notice that the `Option` is parameterized with a tuple type (§3.3.5). This allows for the `apply` method of `Some` to accept a tuple of arguments while accepting a single argument only.

[19] Notice that indexing parameters are not present. These are used for type definitions only, and are not intended to be used in general expressions like instance creation.

[20] Because the implied `apply` method creates new objects and abstract classes can't have any instances of themselves.

A method named `copy` is implicitly added to every case class, unless the case class already has a matching one, or the class has a repeated parameter. The method is defined as follows:

```
def copy [tps] [<uomps>] (ps'_1)...(ps'_n): c[tps] [<uomps>] :=
    c[Ts] [<UoMs>].new(xs_1)...(xs_n)
```

Ts stands for the vector of types defined in the type parameter section *tps* and *UoMs* stands for the vector of units of measure arguments defined in the units of measure parameter section *uomps*. Each xs_i denotes the parameter names of the parameter section ps'_i. Each value parameter of the first parameter list ps'_1 has the form `:` $x_{1,j}$ `:` $T_{1,j}$ `:=` **`self.`** $x_{1,j}$, and the other parameters $ps'_{i,j}$ of the clone constructor are defined as `:` $x_{1,j}$ `:` $T_{1,j}$. Note that these parameters are defined as named parameters (§4.7.7), and that the parameters of the first value parameter section have default values using **`self`**, which is available, since the copied object is already "complete".

A clone constructor is also implicitly added to every case class, but limited to the first value parameter section. The following definition of the implicitly added clone constructor shares the definition of parameters:

```
clone [tps] [<uomps>] (ps'_1): c[tps] [<uomps>]
  self[cloned].@xs_1,1 := xs_1,1
  ...
  self[cloned].@xs_1,n := xs_1,n
end clone
```

Here, $xs_{1,n}$ denotes the (named) parameter names of the first (and only) parameter section.

Every case class implicitly overrides some method definitions of class `Object`, unless a definition of the same method is already given in the case class itself or a concrete definition of the same method is given in some base class (including traits) of the case class, different from `Object`. Namely:

> Method `equals: (Object) ↦ Boolean` (equivalent to operator "=") is structural equality, where two instances are equal if they both belong to the case class in question and they have equal (with respect to `equals`) elements.
>
> Method `hash_code: () ↦ Number.Integer_Unsigned` computes a hash code.
>
> Method `to_string: () ↦ String` returns a string representation which contains the name of the case class and its elements.

5.3.4 Traits

Syntax:

```
Tmpl_Def       ::= 'trait' Trait_Def
Trait_Def      ::= constant_id [Type_Param_Clause] [UoM_Params]
                   Trait_Tmpl_Env
Trait_Tmpl_Env ::= 'extends' [Early_Defs] Trait_Parents
                   semi [Template_Body] 'end' ['trait']
                 | ['extends' [Early_Defs]]
                   '{' [Template_Body] '}'
                 | ['begin' [Template_Body]] 'end' ['trait']
```

A trait is a class that is meant to be injected into some other class as a mixin (including another traits). Unlike normal classes, traits can not be instantiated alone.

Assume a trait D defines some aspect of an instance x of type C (i.e. D is a base class of C). Then the *actual supertype* of D in x is the compound type consisting of all the base classes in $\mathscr{L}(C)$ that succeed D. The actual super type gives the context for resolving a **super** reference in a trait (§6.4). Note that the actual supertype depends on the type to which the trait is added in a trait composition; it is not statically known at the time the trait is defined (the trait must exist before being added anywhere).

If D is not a trait, then its actual supertype is simply its least proper supertype (which is statically known).

Example 5.3.8 The following trait defines the property of being comparable to objects of some type. It contains an abstract operator < and default implementations of the other comparison operators <=, > and >=. Operators are methods, too. The trait also requires the self-type to be $T.

```
trait Comparable[$T <: Comparable[$T]] {
  requires $T
  operator < (that: $T): Boolean end
  operator <=(that: $T): Boolean := self < that || self = that
  operator > (that: $T): Boolean := that < self
  operator >=(that: $T): Boolean := that <= self
}
```

5.3.5 Refinements

There are two separate branches of refinements in Coral, both are quite similar, but used for different purposes. Refinements are kind of a trait (§5.3.4). The branches are as follows:

First, refinements as part of the type system. Those refinements present only declarations and further type restrictions as part of compound types (§3.3.7).

Second, refinements that are basically traits designed to be locally prepended to classes. The second branch is described here.

Syntax:

```
Tmpl_Def            ::= 'refinement' Refinement_Def
Refinement_Def      ::= constant_id 'refine' constant_id
                        Refinement_Tmpl_Env
Refinement_Tmpl_Env ::= '{' [Template_Body] '}'
                      | (semi | 'begin') [Template_Body]
                        'end' ['refinement']
```

Such a refinement does not declare any type, indexing or units of measure parameters, since those are already declared on the type that it refines. Therefore, the type and units of measure parameters are made available to the refinement in order to allow it to override the class methods type-correctly.

Refinements need to be "activated" in the scope for it to take effect (§6.5).

If a refinement refines a parameterized type, then the refinement only activates for this parameterized type (§3.3.4). A refinement can also refine a constrained type (§5.8.2), and the refinement then again only applies to values that conform to this constrained type.

A refinement is not allowed to include or prepend any new traits to the type that it refines. It is an error if it attempts to do so.

A refinement is though allowed to refine classes that are **`final`** or **`sealed`**.

5.3.6 Protocols

Syntax:

```
Tmpl_Def     ::= 'protocol' Pro_Def
Pro_Def      ::= constant_id [Type_Param_Clause] [UoM_Params]
                 Pro_Tmpl_Env
Pro_Tmpl_Env ::= 'extends' Trait_Parents
                 semi [Template_Body] 'end' ['trait']
               | ['extends'] '{' [Template_Body] '}'
               | ['begin' [Template_Body]] 'end' ['trait']
```

Protocols express the contracts that other classes have to implement, and are added to classes with the keyword "**`implements`**".

Protocols are basically traits (§5.3.4) that are stripped of some features, namely:

- Only declarations are allowed as the template body. If a method definition is needed, use a trait instead.
- Protocols can't declare anything for the class or metaclass of the class that implements them. If this is needed, use a trait instead.

- Protocols don't have early definitions (§5.1.8). If this is needed, again, use a trait instead.

Protocol references are preferred to traits to be used by library designers to declare contracts for the code that uses them.

5.3.7 Interfaces

Syntax:

```
Tmpl_Def      ::= ['case'] 'interface' [Ifc_Qualifier] Ifc_Def
Ifc_Qualifier ::= '[' Ifc_Kind ']'
Ifc_Kind      ::= 'class' | 'trait' | 'object'
Ifc_Def       ::= constant_id [Type_Param_Clause]
                  [Dep_Params] [UoM_Params]
                  [Class_Param_Clauses] Ifc_Tmpl_Env
Ifc_Tmpl_Env  ::= 'extends' Class_Parents
                  semi [Template_Body] 'end' ['interface']
                | ['extends'] '{' [Template_Body '}'
                | ['begin' [Template_Body]] 'end' ['interface']
```

Interfaces are filtered versions of classes and traits with only declarations and without early definitions or any definitions at all, except for auxiliary constructor definitions that become empty constructor declarations (without function bodies). Interfaces can be generated from classes or traits by simple transformations and manually edited as needed. Their only purpose is to be used in *module interfaces*, so that implementation is not distributed along, but only declarations in interfaces and protocols are.

5.4 Object Definitions

Syntax:

```
Object_Def ::= constant_id (Class_Template | Trait_Template)
```

Object definitions define singleton instances. If no superclass is given, `Object` is implied, unless the object definition has the same name as an existing or enclosing class – then `Class[`C`]` is implied and the object definition is called a *class object*. If the class definition is not connected to a class, then rules from compound types apply (§3.3.7), the object definition is simply called a *singleton object definition* and is a single object of a new (anonymous) class. This is unlike in Scala, where objects are defined as terms, in a scope separate from types: Coral has one scope for both types and terms.

It's most general form is **`object`** m **`extends`** t. Here, m is the name of the object to be defined (or of the class object, which is the same as the related class), and t is a template (§5.1) of one of the following forms:

Singleton object form. This is a form of objects that are not class objects (not instances of `Class[`C`]`).

sc **with** mt_1 **with** ... **with** mt_n { *stats* }

Class object form. This is a form of objects that are class objects, therefore instances of `Class[`C`]`.

mt_1 **with** ... **with** mt_n { *stats* }

Every trait applied to a class object does not affect the associated class itself – here, the class object is the object that inherits features of the included/prepended traits, not the class instances; and the trait is included/prepended in the metaclass of the class object. This also has an implication on self types of traits (§5.3.4) that are to be included in a class object of class C: the self type would have to be `Class[`C`]`, not just C.

5.4.1 Case Objects

Syntax:

```
Tmpl_Def ::= 'case' 'object' Object_Def
```

Case objects are pretty much similar to case classes (§5.3.3). Case objects can not be class objects at the same time. Their template (§5.1) is the singleton object form of an object definition (§5.4).

5.5 Module Definitions

Syntax:

```
Module_Def  ::= constant_id [Vendor_Arg] {'.' constant_id
                [Vendor_Arg]} [semi] {Module_Expr}
Module_Expr ::= Expr
```

Module definitions are objects that have one main purpose: to join related code and separate it from the outside. Coral's approach to modules solves these issues:

- *Namespaces*. A class with a name C may appear in a module M or a module N, or any other module, and yet be a different object. Modules may be nested.
- *Vendor packages*. Even modules of the same name may co-exists, provided that they have a different vendor, which is just an identifier that looks like a reverse domain name (similar to Java or Scala packages).

- *Dependencies*. Module may define a tree of dependencies, including module vendor resolution, if a module of the same name is provided by different vendors.

Modifiers (§5.2) are available in the same way as in class definitions. This time, members may be classes and other types as well, beside functions.

5.6 Unions

Syntax:

```
Const_Type_Def ::= constant_id 'is' 'union' 'of'
                   '(' Type {semi Type} ')'
```

Union types represent multiple types, possibly unrelated. Union types are abstract by nature and can not be instantiated, only the types that they contain may, if these are instantiable. For type safety, bindings of union types should be matched for the actual type prior to usage.

5.7 Enums

Syntax:

```
Const_Type_Def ::= constant_id [Superclass] 'is' ['bitfield']
                   'enum' '(' Enum_Field {semi Enum_Field} ')'
Enum_Field     ::= constant_id [':=' scalar_literal]
```

Enums (short for Enumerations) are types that contain constants. Bitfield enums may be combined to still produce a single enum value. Every enum constant is a singleton instance of the enum class.

5.8 Dependent Type Declarations

Dependent types consist of three kinds of types in Coral. First, there are *indexed types* (§5.8.1), that are at the core of dependent types. Not every type in Coral is indexed. Secondly, there are two types that are similar and sometimes interchangeable: *constrained types* (§5.8.2) and *range types* (§5.8.3), each making use of indexed types in a specific way. Arguments applied to these types are then described in §3.3.11.

Dependent types are neither concrete nor abstract. They only add a way of indexing existing types and defining subsets of all instances. As a side-effect of this restriction,

variables involved in dependent type declarations are not involved in the rest of the class declarations and definitions, and for that reason don't need to be distinguished from other variables.

Dependent types (§3.3.11) are allowed in these positions:

1. Function parameters.
2. Function return types.
3. Variable declarations and definitions.
4. Type conversions.

5.8.1 Indexed Types

Syntax:

```
Dep_Params         ::= '@[' Index_Param {',' Index_Param} ']'
Index_Param        ::= variable_id [':' Simple_Type]
Indexed_Class_Expr ::= Class_Expr
                     | Indexed_By_Clause
Indexed_By_Clause  ::= 'indexed-with' Indexing_Expr
Indexing_Expr      ::= '{|' Index_Param {',' Index_Param} '|'
                         [Index_Exprs] '}'
                     | '(' Index_Param {',' Index_Param} ')' '->'
                         '{' [Index_Exprs] '}'
Index_Exprs        ::= Index_Expr {[semi] Index_Expr}
Index_Expr         ::= Index_Var Index_Op Index_Val
                     | Index_Val [Index_Op Index_Var]
                     | '(' Index_Var Index_Op Index_Var ')'
                     | '(' Index_Expr [Index_Con Index_Expr] ')'
Index_Var          ::= variable_id
                     | ivar_id
                     | 'self'
Index_Val          ::= variable_id
                     | ivar_id
                     | literal
                     | constant_id
                     | Index_Fun '(' Index_Var {',' Index_Var} ')'
```

Indexed types declare what their index is, based on combinations of their input (indexing) variables, instance variables and operators and functions working with these. Since the scope of testing these *indexing constraints* is limited to cases listed in §5.8, independent on the intrinsic state of the instances, the instances may appear in states that do not conform to the constraints in between each indexing constraint test, but to

pass as the dependent type, they must conform to the indexing constraint at the time the test is invoked, that is:

1. Method resolution time (function parameters).
2. Returning from a function (function return types).
3. Getting assigned to a variable.
4. Getting converted to a type (unless the conversion is implicit to a different type).

Note that implicit conversions (§6.26.2) can't apply to dependent types, since that would be a conversion from the same type to a subset of the same type.

Sorts. This part of Coral is inspired by the ATS language. "Sorts" are a types for which the language knows ordering of their values implicitly:

- `Boolean`.
- `Integer`.
- `Float`.
- `Char`.
- Every **`enum`** type (§5.7), where the ordering is given by the order in which each enumerated value appears.
- Every type constrained from a pre-existing sort (§5.8.2).

See references of these "sort" types for more details on their particular ordering.

"Sorts" are the types allowed as types of *indexing variables* (see `Index_Param` syntax). If no type is specified, `Integer` "sort" is implied.

For the mentioned reasons, `Dep_Sort_Val` (used in §3.3.11) are values that are members of "sorts".

Note that the `Index_Exprs` are optional – meaning that the entire "sort" is used for indexing, not only a subset of it.

Indexing Operators. Indexing operators used to declare indexing constraints are the following for the `Index_Op` syntax:

- ~ (bitwise negation): `(Number)` ↦ `Number`
- + (addition): `(Number, Number)` ↦ `Number`
- – (subtraction): `(Number, Number)` ↦ `Number`
- * (multiplication): `(Number, Number)` ↦ `Number`
- ** (exponentiation): `(Number, Number)` ↦ `Number`
- / (division): `(Number, Number)` ↦ `Number`
- % (modulo): `(Number, Number)` ↦ `Number`
- > (greater than): `(Number, Number)` ↦ `Boolean`
- >= (greater than or equal to): `(Number, Number)` ↦ `Boolean`
- < (less than): `(Number, Number)` ↦ `Boolean`
- <= (less than or equal to): `(Number, Number)` ↦ `Boolean`
- = (equal to): `(Number, Number)` ↦ `Boolean`)
- <> (not equal to): `(Number, Number)` ↦ `Boolean`
- != (not equal to): `(Number, Number)` ↦ `Boolean`
- ! (boolean negation): `(Boolean)` ↦ `Boolean`
- || (boolean disjunction): `(Boolean, Boolean)` ↦ `Boolean`
- && (boolean conjunction): `(Boolean, Boolean)` ↦ `Boolean`
- ^^ (boolean exclusive disjunction): `(Boolean, Boolean)` ↦ `Boolean`
- | (bitwise or): `(Number, Number)` ↦ `Number`
- & (bitwise and): `(Number, Number)` ↦ `Number`
- ^ (bitwise xor): `(Number, Number)` ↦ `Number`

These operators are static[21]. In the list, `Number` can also be replaced by `Char`, but not by `Complex`.

[21]That is, not overridable by user programs.

Indexing Functions. Functions that can be used to constrain the indexing are also static and limited to work only with a restricted number of types:

- `size: (List) ↦ Number`
- `length: (String) ↦ Number`
- `empty?: (List) ↦ Boolean`
- `empty?: (String) ↦ Boolean`
- `max: (*Number) ↦ Number`
- `min: (*Number) ↦ Number`
- `avg: (*Number) ↦ Number`
- `sum: (*Number) ↦ Number`
- `abs: (*Number) ↦ Number`
- `sgn: (Number) ↦ Number`
- `log: (Number, Number) ↦ Number`
- `ln: (Number) ↦ Number`
- `even?: (Number) ↦ Boolean`
- `odd?: (Number) ↦ Boolean`
- `concat: (*String) ↦ String`
- `coalesce: (*Object) ↦ Object`
- `lowercase: (String) ↦ String`
- `uppercase: (String) ↦ String`

Indexing Constraint Concatenation. If there are multiple indexing constraints separated by the `[semi]` syntax, boolean conjunction is implied. Indexing constraints may be joined by a different boolean operations (`Index_Con`):

- `||` (boolean disjunction): `(Boolean, Boolean) ↦ Boolean`
- && (boolean conjunction): `(Boolean, Boolean) ↦ Boolean`
- ^^ (boolean exclusive disjunction): `(Boolean, Boolean) ↦ Boolean`

Example 5.8.1 The following is an example on how a `String` type might index itself:

```
class String @[length: Integer]
  indexed-with (length) -> {@length = length}
end
```

5.8.2 Constrained Types

Syntax:

```
Type_Def      ::= constant_id [Type_Param_Clause] ':='
                  Type [Dep_Params_C]
Dep_Params_C  ::= '@[|' Index_Param {',' Index_Param} '|'
                  [Index_Exprs_C] ']'
                | '@(' Index_Param {',' Index_Param} ')'
                  '->' '{' [Index_Exprs_C] '}'
Index_Exprs_C ::= Index_Expr_C {[semi] Index_Expr_C}
Index_Expr_C  ::= Index_Var_C Index_Op Index_Val_C
                | Index_Val_C [Index_Op Index_Var_C]
                | '(' Index_Var_C Index_Op Index_Var_C ')'
                | '(' Index_Expr_C
                  [Index_Con Index_Expr_C] ')'
Index_Var_C   ::= variable_id
Index_Val_C   ::= variable_id
                | literal
                | constant_id
                | Index_Fun '(' Index_Var_C
                  {',' Index_Var_C} ')'
```

Constrained types are basically aliases to indexed types with optionally further restricted indexing, which can only make use of the existing indexing constraints of the base type, using the indexing variables. No additional instance variable can be constrained by a constrained type directly.

The type of the indexing variable is implied to be the same as of the indexed type.

Example 5.8.2 Here is an example of how Coral might declare the `Char` type:

```
type Char := String @[|length| length = 1]
type Char := String @(length) -> {length = 1}
```

Notice how the constraint only uses the already existing indexing variable.

Example 5.8.3 Here is an example of a different constrained type:

```
type Even_Positive_Integers ::= Integer @[|i| even?(i); i >= 0]
type Even_Positive_Integers ::= Integer @(i) -> {even?(i); i >= 0}
```

This constrained type declares `Even_Positive_Ints` to be positive integers that are even at the same time.

5.8.3 Range Types

Syntax:

```
Const_Type_Def ::= constant_id 'is' 'range'
                     (Range_Expr
                   | '(' Range_Expr ')' [':' constant_id])
```

Range types are similar to constrained types, but limited in a few ways: they can constrain only indexed types that are indexed with exactly one indexing variable. The range expression is converted into the corresponding indexing constraint.

Example 5.8.4 An example of a constrained type interchangeable with a range type:

```
type Positive_Integers ::= Integer @(i) -> {i >= 0}
type Positive_Integers is range
  (0 .. +Integer.Infinity) : Integer
end type
type Positive_Integers is range
  0 .. +Integer.Infinity
end type
```

Types that do not require the "`constant_id`" are those that are "sorts" at the same time, so that the indexed type can be inferred.

5.9 Units of Measure

Syntax:

```
Const_Type_Def ::= Unit_Name 'is' ['abstract'] 'unit-of-measure'
                     [semi Unit_Convs {semi Unit_Convs}]
Unit_Name      ::= variable_id | constant_id
                     ['extends' Superunit_Name]
Superunit_Name ::= variable_id | constant_id
Unit_Convs     ::= Unit_Name ':=' Unit_Conv
Unit_Conv      ::= '(' Unit_Conv ')'
                   | Unit_Elem [Unit_Op Unit_Elem]
                   | Unit_Conv Unit_Op Unit_Conv
```

```
Unit_Elem      ::= number_literal | Unit_Name
Unit_Op        ::= '*' | '/' | '**'
```

Numbers in Coral can have associated units of measure, which are typically used to indicate length, volume, mass, distance and so on. By using quantities with units, the runtime is allowed to verify that arithmetic relationships have the correct units, which helps prevent programming errors.

Example 5.9.1 The following defines the measure `cm` (centimeter).

```
type cm is unit-of-measure
end type
```

Example 5.9.2 The following defines the measure `ml` (milliliter) as a cubic centimeter (`cm ** 3`).

```
type ml is unit-of-measure
  ml := cm ** 3
end type
```

Example 5.9.3 The following shows possible usage of abstract units of measure.

```
type distance is abstract unit-of-measure
end type

type m extends distance is unit-of-measure
  m := km / 1000
end type

type km extends distance is unit-of-measure
  km := m * 1000
end type

type mi extends distance is unit-of-measure
  mi := km * 0.621
  mi := (m * 1000) * 0.621 // this can be inferred!
end type
```

This enables types aggregated with units of measure require a number tagged with any distance unit of measure and still work with correct units.

Every unit of measure is defined in the same scope as any other type would be, but the application of units of measure to numbers or *aggregated unit types* require to import units of measure by name into the scope where a unit of measure from a different unrelated module would be used.

Types Aggregated with Units of Measure. In addition to type parameters and dependency parameters of each type, every type may be parameterized with a units of measure aggregation. It is recommended to avoid mixing these three together.

Syntax:

```
UoM_Params ::= '[<' UoM_Param {',' UoM_Param} '>]'
UoM_Param  ::= '$' Unit_Name
```

Names of unit of measure parameters must not clash with names of type parameters, otherwise it is a compile-time error.

Persistence of Units of Measure. There is a huge difference between the way F# handles units of measure and Coral's way. In F#, the unit of measure information is lost after compilation, but persists in Coral in runtime, since verification of units of measure is deferred also to runtime, as it is limited during compilation. This also means that the information may be accessed in runtime, e.g. using it to print the unit information on screen.

5.10 Record Types

Syntax:

```
Const_Type_Def    ::= Record_Name 'is' ['abstract'] 'record'
                      [semi] Record_Components
Record_Name       ::= constant_id [Type_Param_Clause] [Param_Clause]
Record_Components ::= Record_Component {semi Record_Component}
                      [Superclass]
Record_Component  ::= 'val' Val_Dcl
                    | 'var' Var_Dcl
                    | 'case' variable_id semi
                      {'when' constant_id
                      ('then' | nl)
                      Record_Components} 'end' ['case']
```

Record types are simple syntax sugar for classes that represent *data objects*, i.e., objects that don't really care about behaviour, their only purpose is to store data.

Record types can appear in three different forms: *basic records*, *discriminated records* and *variant records*.[22]

[22]Note that the syntax for record types in Coral differs from Ada's `type Record_Name is record... end record;`. Coral ends a type, not a record. This difference appears in more places than just this syntax.

Basic Records. Basic records have no discriminating parameters, and can have only type parameters. Their structure is always the same.

Discriminated Records. Discriminated records are similar to variant records, they can have discriminating parameters and a discriminated record may also be a variant record. Discriminating parameters are to be used in conjunction with dependent types (§5.8) of the record's components. Discriminating a record type does the same thing as declaring a dependent type: it creates a subset of instances of the original record type.

Variant Records. Variant records are similar to discriminated records, they can have discriminating parameters and a discriminated record may also be a discriminated record. Variant parameters are enums (§5.7). Variant records render new subtypes of the original record type, similar to a concrete type constructor.

Example 5.10.1 The following example defines a variant record.

```
type Traffic_Light is bitfield enum
  Red
  Yellow
  Green
end type

type Variant_Record (option: Traffic_Light) is record
  val a: A
  var b: B
  case option
  when Red
    val c: C
  when Yellow
    var d: D
  when Green
    val e: E
  end case
end type

let vr := Variant_Record(Traffic_Light.Red).
  new(a: A.new, b: B.new, c: C.new)
```

5.11 Nullability

Every type in Coral conforms to `Object`. In turn, `Nothing` conforms to any type. But, the only instance of `Nothing`, which is a singleton accessed with the keyword **`nil`**, can

not be assigned to every typed variable. If a type is declared as nullable, then it can. If a type is declared as not-nullable, which is the implicit preference for every type, then it can not.[23]

The implicit preference may be changed in two levels. These levels are *preferred nullability* and *explicit nullability* (§3.3.10).

Preferred nullability is switched by annotations on each class: `Nullable`. The annotation has one positional parameter of type `Boolean` defining what the nullability will be (**`yes`** for nullable types, **`no`** for not-nullable types), and a named parameter `:override` of type `Boolean`, defining whether subclasses may override this preference, and implicitly set to **`yes`**, meaning that subclasses may override this preference by default.

All types in Coral are not-nullable implicitly to prevent `Method_Not_Found_Errors`, resulting from sending messages to **`nil`**. But it is clear that for some cases, nullability of types may be desired, such as with dictionaries, so that **`nil`** can be returned for keys that have no value in the dictionary. However, it is preferred to use the `Option` type wherever possible to indicate that the value may not be present, and moreover, the `Option` type can be used in pattern matching even for values that are not of the `Option` type, utilizing an extractor pattern (§8.1.8) with `Some` and a constant pattern (§**??**) with `None`, instead of constructor patterns (§8.1.6). This is again different from Scala, where `Some` and `None` can't be used to match any value, only `Option` values. This difference in Coral makes nullable types and the `Option` type interoperable in pattern matching.

Example 5.11.1 Here are all four versions of the nullability annotation.

```
@[Nullable yes, override: yes]
@[Nullable yes, override: no]
@[Nullable no, override: yes]
@[Nullable no, override: no]
```

[23] An original idea was to make every type not-nullable and force users to use the `Option` type for every no-object scenario. But then, what would **`nil`** be good for?

Chapter 6

Expressions

Syntax:

```
Expr        ::= Cond_Expr
              | Loop_Expr
              | Rescue_Expr
              | Raise_Expr
              | Throw_Expr
              | Catch_Expr
              | Return_Expr
              | Assign_Expr
              | Update_Expr
              | Yield_Expr
              | Infix_Expr
              | Simple_Expr
              | Match_Expr
              | Binding
              | Annot_Expr
              | Cast_Expr
              | Use_Expr
              | Jump_Expr
              | Anon_Fun
              | Metaclass_Access
Infix_Expr  ::= Prefix_Expr
              | Infix_Expr [op_id Infix_Expr]
Simple_Expr ::= '(' Anon_Class ')'
              | Block_Expr
              | Simple_Expr1 ['_']
Result_Expr ::= Anon_Params '->' Block
              | ['memoize'] Expr
```

Expressions are composed of various keywords, operators and operands. Expression

forms are discussed subsequently.

6.1 Expression Typing

The typing of expressions is often relative to some *expected type* (which might be undefined). When we write "expression e is expected to conform to type T", we mean:

1. The expected type of e is T.
2. The type of expression e must conform to T.

Usually, the type of the expression is defined by the last element of an execution branch, as discussed subsequently with each expression kind.

What we call "statement", in context of Coral is in fact yet another kind of an expression, and those expressions themselves always have a type and a value.

6.2 Literals

Syntax:

```
Simple_Expr1 ::= literal
```

Typing of literals is as described in (§1.5); their evaluation is immediate, including non-scalar literals (collection literals).

6.2.1 The Nil Value

Syntax:

```
Simple_Expr1 ::= 'nil'
```

The **`nil`** value is of type `Nothing`, and is thus compatible with every type that is nullable (§5.11), either preferably or explicitly.

The **`nil`** represents a "no object", and is itself represented by an object. This object overrides methods in `Object` as follows:

- `equals(`x`)` and `=(`x`)` return **`yes`** if the argument x is also the **`nil`** object.
- `!=(`x`)` return **`yes`** if the argument x is not the **`nil`** object.
- `as_instance_of[$T]()` returns always **`nil`**.

- `hash_code()` returns `0`.

A reference to any other member of the **`nil`** object causes `Method_Not_Found_Error` or `Member_Not_Found_Error` to be raised, unless the member in fact exists.[1]

6.3 Designators

Syntax:

```
Simple_Expr1 ::= Path
               | Simple_Expr '.' importable_id
```

A designator refers to a named term. It can be a *simple name* or a *selection*.

6.4 Self, This & Super

Syntax:

```
Simple_Expr1 ::= [Container_Path '.'] 'self'
                 ['.' (constant_id | variable_id | function_id)]
               | [Container_Path '.'] 'this'
                 ['.' (constant_id | variable_id | function_id)]
               | [Container_Path '.'] 'super'
                 [Class_Qualifier]
                 ['.' (constant_id | variable_id | function_id)]
               | [Container_Path '.'] 'self' '[' 'cloned' ']'
                 '.' ivar_id
               | ivar_id
               | cvar_id
```

The expression **`self`** stands always for the current instance in the context and function resolution searches in the actual class of the instance.

The expression **`this`** is the same as **`self`**, except that function resolution searches from the class that this expression appears in, possibly skipping overrides.

6.5 Use Expressions

Syntax:

[1] It is even possible to use a refinement to actually implement some methods of **`nil`** locally (preferred approach), or globally implement those methods (discouraged, causes warnings).

```
Use_Expr ::= 'use' Simple_Expr 'as'
             [variable_id ':'] Type [Block_Expr]
```

Use expressions are similar to type cast expressions. Their intention is to rebind an expression to a specific type, and then either have this type be effective in the same scope from that point onward, or, if a `Block_Expr` is syntactically given, only in the scope of that block expression. If a block is given, then the return value of the block is the value of this expression, otherwise, the value retrieved by evaluation of `Simple_Expr` is the value of this expression.

6.6 Function Applications

Syntax:

```
Simple_Expr1   ::= Simple_Expr1 Argument_Exprs
Argument_Exprs ::= '(' [Arg_Exprs] ')' [Block_Expr]
                 | '(' [Arg_Exprs ','] '*' Expr ')' [Block_Expr]
                 | '(' [Arg_Exprs ','] ['*' Expr ','] '**' Expr ')'
                   [Block_Expr]
                 | Arg_Exprs [Block_Expr2]
                 | [Arg_Exprs ','] '*' Expr [Block_Expr2]
                 | [Arg_Exprs ','] ['*' Expr ','] '**' Expr
                   [Block_Expr2]
                 | Block_Expr
Arg_Exprs      ::= Arg_Expr {',' Arg_Expr}
Arg_Expr       ::= Expr
```

6.6.1 Named and Optional Arguments

6.6.2 By-Name Arguments

6.6.3 Input & Output Arguments

6.6.4 Function Compositions & Pipelines

6.7 Type Applications

Syntax:

```
Simple_Expr1 ::= Simple_Expr (Type_Args [UoM_Args] | UoM_Args)
```

6.8 Tuples

Syntax:

```
Simple_Expr ::= '(' [Exprs] ')'
```

A tuple expression `(`$e_1, \ldots, e_n$`)` is an alias for the class instance creation `Tuple_`n`(`$e_1, \ldots, e_n$`)`, where $n \geq 2$. The empty tuple `()` is the unique value of type `Unit`. A tuple with only one value is only the value itself, without being wrapped in a tuple.

6.9 Instance Creation Expressions

Unlike languages like Java, Scala, C# and similar, Coral does not have dedicated language construct for creating new instances of classes. Instead, all such attempts are made through the `Class#new` method (not to be confused with `Class.new`), which in the end[2] has all arguments for a constructor, which gets invoked by a native implementation.

6.10 Blocks

Syntax:

```
Block_Expr  ::= Block_Expr1 | Block_Expr2
Block_Expr1 ::= '{' [Block_Args semi] Block '}'
Block_Expr2 ::= 'do' [Block_Args semi] Block 'end'
Block_Args  ::= '|' [Params] '|' [':' Type]
Block       ::= {Block_Stat semi} [Result_Expr]
```

6.11 Yield Expressions

Syntax:

```
Yield_Expr ::= 'yield' '(' [Arg_Exprs] ')'
             | '(' [Arg_Exprs ','] '*' Expr ')'
             | '(' [Arg_Exprs ','] ['*' Expr ','] '**' Expr ')'
```

[2]Because constructor currying can happen, the constructor is invoked by a native implementation from outside of the `new` method, by the curried function, which is a mechanism inaccessible to users otherwise than via the constructor definition.

```
              | [Arg_Exprs]
              | [Arg_Exprs ','] '*' Expr
              | [Arg_Exprs ','] ['*' Expr ','] '**' Expr
```

6.12 Prefix & Infix Operations

Syntax:

```
Infix_Expr  ::= Prefix_Expr
              | Infix_Expr [op_id Infix_expr]
Prefix_Expr ::= [op_id] Simple_Expr
```

6.12.1 Prefix Operations

6.12.2 Infix Operations

6.12.3 Assignment Operators

6.13 Typed Expressions

Syntax:

```
Cast_Expr ::= Infix_Expr 'as' Type
```

6.14 Annotated Expressions

Syntax:

```
Annot_Expr ::= Annotation {Annotation} Infix_Expr
```

6.15 Assignments

Syntax:

```
Assign_Expr ::= [Simple_Expr '.'] variable_id ':=' Expr
Update_Expr ::= Simple_Expr1 Argument_Exprs ':=' Expr
```

6.16 Conditional Expressions

Syntax:

```
Cond_Expr        ::= Cond_Block_Expr | Cond_Mod_Expr
Cond_Block_Expr  ::= Cond_Block_Expr1 | Cond_Block_Expr2
Cond_Block_Expr1 ::= 'if' Expr ('then' | semi) Expr
                     {'elsif' Expr ('then' | semi) Expr}
                     ['else' Expr] 'end' ['if']
Cond_Block_Expr2 ::= 'unless' Expr ('then' | semi) Expr
                     {'elsif' Expr ('then' | semi) Expr}
                     ['else' Expr] 'end' ['unless']
Cond_Mod_Expr    ::= Expr Cond_Modifier
Cond_Modifier    ::= Cond_Modifier1
                     ['else' Infix_Expr]
Cond_Modifier1   ::= ('if' | 'unless') Infix_Expr
```

6.17 Loop Expressions

6.17.1 Iterable For Expressions

Syntax:

```
Loop_Expr       ::= [Label_Dcl] 'for' Val_Dcl 'in' ['reverse'] Expr
                    ['step' Expr] For_Loop
For_Loop        ::= 'loop' Expr 'end' ['loop']
                  | '{' Loop_Block_Expr '}'
Loop_Block_Expr ::= {Block_Stat | Loop_Ctrl_Expr}
```

6.17.2 Loop Control Expressions

Syntax:

```
Loop_Ctrl_Expr ::= Break_Expr
                 | Skip_Expr
                 | Next_Expr
Break_Expr     ::= 'break' [label_id] [Cond_Modifier1]
Skip_Expr      ::= 'skip' [integer_literal] [Cond_Modifier1]
Next_Expr      ::= 'next' [label_id] [Cond_Modifier1]
```

6.17.3 While & Until Loop Expressions

Syntax:

```
Loop_Expr     ::= [Label_Dcl] ('while' | 'until') Expr For_Loop
                | Loop_Mod_Expr
Loop_Mod_Expr ::= Expr Loop_Modifier
Loop_Modifier ::= ('while' | 'until') Expr
```

6.17.4 Pure Loops

Syntax:

```
Loop_Expr  ::= [Label_Dcl] 'loop'
               (semi Loop_Block_Expr 'end' ['loop'] |
               '{' Loop_Block_Expr '}')
```

6.18 Generator Expressions

Syntax:

```
Loop_Expr      ::= 'for' (Generator_Iter | Generator_Expr)
Generator_Iter ::= '(' Enumerators ')' (Expr | For_Loop)
Generator_Expr ::= '{' Enumerators semi 'yield' Expr '}'
Enumerators    ::= Generator {semi Enumerator}
Enumerator     ::= Generator
                 | Enumerator_Guard
                 | Pattern1 ':=' Expr
Generator      ::= Pattern1 '<-' Expr [Guard]
Guard          ::= Cond_Modifier1
```

6.19 Collection Comprehensions

Syntax:

```
List_Literal       ::= '%' Collection_Flags '[' 'for' Generator_Expr ']'
Dictionary_Literal ::= '%' Collection_Flags '{' 'for' Generator_Expr '}'
Bag_Literal        ::= '%' Collection_Flags '(' 'for' Generator_Expr ')'
```

Collection comprehensions extend the syntax of collection "literals"[3], so that collections may be defined by not their explicit values, but by a function that generates them – and that function is a generator. Only tuple literals don't have collection comprehension, due to their special nature within the language.

6.20 Pattern Matching & Case Expressions

Syntax:

```
Match_Expr     ::= Pat_Match_Expr | Case_Expr
Pat_Match_Expr ::= 'match' Simple_Expr1 Match_Body
Match_Body     ::= semi When_Clauses 'end' ['match']
                 | '{' When_Clauses '}'
When_Clauses   ::= When_Clause {['next'] semi When_Clause}
                   [['next'] 'else' Block]
When_Clause    ::= 'when' Pattern [Guard] ('then' | semi) Block
Case_Expr      ::= 'case' Simple_Expr1 Case_Body
Case_Body      ::= semi Case_Clauses 'end' ['case']
                 | '{' Case_Clauses '}'
Case_Clauses   ::= Case_Clause {['next'] semi Case_Clause}
                   [['next'] 'else' Block]
Case_Clause    ::= 'when' Case_Patterns ('then' | semi) Block
Case_Patterns  ::= Case_Pattern {',' Case_Pattern}
Case_Pattern   ::= Stable_Id
                 | literal [('..' | '...') literal]
                 | variable_id
                 | Infix_Expr
```

6.21 Unconditional Expressions

Unconditional expressions change the flow of programs without a condition.

6.21.1 Return Expressions

Implicit Return Expressions. Implicit return expression is always the value of the last expression in a code execution path.

Syntax:

[3]Pure literals are terminal symbols in the language, but collection literals are wrappers around virtually any expression.

```
Result_Expr ::= Anon_Params '->' Block
              | ['memoize'] Expr
```

Explicit Return Expressions. Explicit return expressions unconditionally change the flow of programs by making the enclosing function definition return a value early (or return no value).

Syntax:

```
Return_Expr ::= ['memoize'] 'return' [Expr] [Cond_Modifier1]
```

6.21.2 Structured Return Expressions

Syntax:

```
Return_Expr ::= ['memoize'] 'return' Var_Def 'do'
                Block_Stat {semi Block_Stat}
                'end' ['return']
```

6.21.3 Local Jump Expressions

Syntax:

```
Jump_Expr  ::= Goto_Expr | Label_Dcl
Goto_Expr  ::= 'goto' label_name [Cond_Modifier1]
Label_Dcl  ::= 'label' label_name
             | '<<' label_name '>>'
label_name ::= variable_id | constant_id
```

Local jumps transfer control from **`goto`** statements to the statements following a **`label`**.

6.21.4 Continuations

Unlimited continuations. Unlimited continuations are defined by the whole program, as the unlimited continuation allows almost arbitrary non-local jumps. The unlimited continuation is captured with `call/cc` function.

Delimited continuations. Delimited continuations are defined with `reset` and `shift` functions. Reset and shift expressions are actually not language constructs, but rather regular functions that have a native implementation capable of unconditionally changing the standard control flow of a program. Moreover, the first `shift` expression (which captures the delimited continuation) controls the return value of the `reset` expression, which overrides the implicit return expression (§6.21.1).

6.22 Throw & Catch Expressions

Syntax:

```
Catch_Expr  ::= Catch_Expr1 | Catch_Expr2
Catch_Expr1 ::= 'begin' Block
                'catch' semi When_Clauses
                'end'
Catch_Expr2 ::= '{' Block '}'
                'catch' '{' When_Clauses '}'
Throw_Expr  ::= 'throw' Expr
```

6.22.1 Raise Expressions

Syntax:

```
Raise_Expr ::= 'raise' Raiseable
Raiseable  ::= string_literal
             | Container_Path [',' string_literal]
             | Expr
```

6.22.2 Rescue & Ensure Expressions

Syntax:

```
Rescue_Expr  ::= Rescue_Expr1 | Rescue_Expr2
Rescue_Expr1 ::= 'begin' Block
                 {'rescue' [Pattern [Guard]] semi Block }
                 ['ensure' semi Block_Stat {semi Block_Stat}] 'end'
Rescue_Expr2 ::= '{' Block '}'
                 {'rescue' [Pattern [Guard]] '{' Block '}'}
                 ['ensure' '{' Block_Stat {semi Block_Stat} '}']
Fun_Stats    ::= Block
                 {'rescue' [Pattern [Guard]] semi Block }
                 ['ensure' semi Block_Stat {semi Block_Stat}]
```

6.23 Anonymous Functions

Syntax:

```
Anon_Fun        ::= Anon_Params '->' '{' Block '}'
Result_Expr     ::= Anon_Params '->' Block
```

```
Anon_Params     ::= Bindings {'->' Bindings}
                  | Param_Clause
                  | '(' ['implicit'] variable_id ')'
                  | '(' [Nameless_Param] ')'
Bindings        ::= '(' Binding {',' Binding} ')'
Binding         ::= (variable_id | '_') [':' Type]
Nameless_Params ::= Nameless_Param {',' Nameless_Param}
Nameless_Param  ::= '_' [':' Compound_Type]
```

6.23.1 Method Values

Syntax:

```
Simple_Expr ::= Simple_Expr1 '_'
```

Example 6.23.1 The method values in the left column are each equivalent to the anonymous functions (§6.23) on their right.

```
Math.sin _               (x)      -> { Math.sin(x) }
Array.range _            (x1, x2) -> { Array.range(x1, x2) }
List.map_2 _             (x1, x2) -> (x3) -> { List.map_2(x1, x2)(x3) }
List.map_2(xs, ys) _     (x)      -> { List.map_2(xs, xy)(x) }
42.* _                   (x)      -> { 42 * x }
```

6.24 Anonymous Classes

Syntax:

```
Anon_Class      ::= ['class' [Class_Param_Clauses] 'extends']
                    [Early_Defs] Anon_Class_Tmpl
Anon_Class_Tmpl ::= Class_Parents 'with' '{' [Template_Body] '}'
```

Anonymous classs are a mechanism to implement an abstract class or override a concrete class "ad hoc", in place where needed, without needing to create a new constant (although as an expression, the anonymous class definition can indeed be assigned to a constant and gain its name). Anonymous classes can't be type constructors (§3.4.3).

A minimal anonymous class expression is of the form c `with` `{` t `}`, where c is the class that the anonymous class inherits from (can be even `Object`), and t is the template of the anonymous class. The anonymous class inherits all traits mixed into this parent class, and can itself include or prepend more traits (via the `Class_Parents` syntax element).

Optionally, the anonymous class may define its own primary constructor parameters, in which case the form of the anonymous class is **class** $(ps_1)\ldots(ps_n)$ **extends** c **with** { t }, where ps_1 to ps_n are the primary constructor parameters. Superclass constructor arguments may be specified in any case.

6.25 Statements

Syntax:

```
Block_Stat    ::= Use
                | {Annotation} ['implicit'] Def
                | {Annotation} {Local_Modifier} Tmpl_Def
                | Expr
                | ()
Template_Stat ::= Use
                | {Annotation} {Modifier} Def
                | {Annotation} {Modifier} Dcl
                | Expr
                | ()
Fun_Stats     ::= [Fun_Stat {semi Fun_Stat}] Return_Expr
                | Block
                  {'rescue' [Pattern [Guard]] semi Block }
                  ['ensure' semi Block_Stat {semi Block_Stat}]
Fun_Stat      ::= Block_Stat
Fun_Dec_Expr  ::= {Annotation} Dcl
                | {Annotation} ['implicit'] Def
                | 'transparent'
                | 'opaque'
                | 'native' [Expr]
                | ()
```

Statements occur as parts of blocks and templates. Despite their name, they are actually generally expressions as well, except that for some statements, their value is not much of a use, i.e. use clauses, whose value is a **nil**, or the empty statement/expression, whose value is again **nil**.

Function statements is an umbrella term for a series of statements and expressions, so their effective value is more complex.

Statement can be an import via a use clause (§4.9), a definition or an expression, or it can be empty. Statements used in the template of a class definition can also be declarations.

6.26 Conversions

6.26.1 Explicit Conversions

6.26.2 Implicit Conversions

6.27 Workflows

Chapter 7

Implicit Parameters & Views

7.1 The Implicit Modifier

Syntax:

```
Local_Modifier ::= 'implicit'
Param_Clauses  ::= {Param_Clause} '(' 'implicit' Params ')'
```

7.2 Implicit Parameters

An implicit parameter list (**implicit** $p_1, \ldots, p_n$) of a method marks the parameters $p_1, \ldots, p_n$ as implicit.

7.3 Views

7.4 View Bounds

Syntax:

```
Type_Param ::= (tp_id | '_') [Type_Param_Clause]
               ['>:' Type] ['<:' Type] {'<%' Type}
```

Chapter 8

Pattern Matching

8.1 Patterns

Syntax:

```
Pattern        ::= Pattern1 {'|' Pattern1}
Pattern1       ::= variable_id ':' Type_Pat
                 | '_' ':' Type_Pat
                 | Pattern2
Pattern2       ::= variable_id ['@' Simple_Pattern]
                 | Simple_Pattern
Simple_Pattern ::= ['*'] '_'
                 | ['*'] variable_id
                 | literal
                 | Stable_Id
                 | Stable_Id '(' [Patterns] ')'
                 | '(' [Patterns] ')'
                 | Pattern '&' Pattern
                 | Pattern '~>' Pattern
                 | Pattern '<~' Pattern
                 | Pattern ('..' | '...') Pattern
Patterns       ::= Pattern {',' Pattern}
```

8.1.1 Variable Patterns

Syntax:

```
Simple_Pattern ::= '_'
                 | variable_id
```

A variable pattern x is a simple identifier which starts with a lower case letter. It matches any value and binds the variable name to that value. The type of x is the expected type of the pattern as given from the outside. A special case is the wildcard pattern "_", which is treated as if it was a fresh variable on each occurence, and which does not bind itself to the value, i.e., it is alone equivalent to the **else** clause of `When_Clauses`.

8.1.2 Typed Patterns

Syntax:

```
Simple_Pattern ::= '_' ':' Type_Pat
                 | variable_id ':' Type_Pat
Type_Pat       ::= Type
```

A typed pattern x`:` T consists of a pattern variable x and a type pattern T. The type of x is the type T, where each type variable and wildcard is replaced by a fresh, unknown type. This pattern matches any value matched by the type pattern T (§8.2), and it binds the variable name to that value (unless the variable name is "_").

8.1.3 Pattern Binders

Syntax:

```
Pattern2 ::= variable_id '@' Simple_Pattern
```

A pattern binder x `@` p consists of a pattern variable x and a pattern p. The type of the variable x is the type T resulting from the pattern p. This pattern matches any value v matched by the pattern p, provided the type of v is also an instance of T, and it binds the variable name to that value.

Example 8.1.1 In the following example, `person` binds to the whole `Person` object.

```
def f (someone: Person) := match someone
  when person @ Person('John Galt', _, _) then ...
end match
```

8.1.4 Literal Patterns

Syntax:

```
Simple_Pattern ::= literal
```

A literal pattern L matches any value that is equal (in terms of =) to the literal L. The type of L must conform to the expected type of the pattern. Literal kinds that are considered legal with this pattern are: string literals, number literals, boolean literals and the **nil** value.

8.1.5 Stable Identifier Patterns

Syntax:

```
Simple_Pattern ::= Stable_Id
```

A stable identifier pattern is a stable identifier r (§3.2). The type of r must conform to the expected type of the pattern. The pattern matches any value v, such that $r = v$.

To resolve the syntactic overlap with a variable pattern (§8.1.1), a stable identifier pattern may not be a simple name starting with a lower case letter. However, it is possible to enclose such a variable or method name in backquotes, then it is treated as a stable identifier pattern.

Example 8.1.2 Consider the following function definition:

```
def f (x: Integer, y: Integer) := match x
  when y then ...
end match
```

Here, `y` is a variable pattern, which matches any value, namely here it would bind simply to `x`. If we wanted to turn the pattern into a stable identifier pattern, this can be achieved as follows:

```
def f (x: Integer, y: Integer) := match x
  when `y` then ...
end match
```

Now, the pattern matches the `y` parameter of the enclosing function `f`. That is, the match succeeds only if the `x` argument and the `y` argument of `f` are equal.

8.1.6 Constructor Patterns

Syntax:

```
Simple_Pattern ::= Stable_Id '(' [Patterns] ')'
```

A constructor pattern is of the form $c(p_1, \ldots, p_n)$, for $n \geq 0$. It consists of a stable identifier c, followed by element patterns $p_1, \ldots, p_n$. The constructor c is a simple or qualified name which denotes a case class (§5.3.3). If the case class is monomorphic,

then it must conform to the expected type of the pattern, and the formal parameter types of c's primary constructor (§5.3.1) are taken as the expected types of the element patterns $p_1, \ldots, p_n$. If the case class is polymorphic, then its type parameters are instantiated so that the instantiation of c conforms to the expected type of the pattern, unless the type arguments are already given. These types of the formal parameter types of c's primary constructor are then taken as the expected types of the component patterns $p_1, \ldots, p_n$. The pattern matches all objects created from constructor invocations $c(p_1, \ldots, p_n)$, where each element pattern p_i matches the corresponding value v_i. Any extra parameter sections of c's primary constructor do not affect this behavior.

A special case arises when c's formal parameter types contain a repeated parameter. This is further discussed in (§8.1.9).

8.1.7 Tuple Patterns

Syntax:

```
Simple_Pattern ::= '(' [Patterns] ')'
```

A tuple pattern $(p_1, \ldots, p_n)$ is an alias for the constructor pattern `Tuple_`$n(p_1, \ldots, p_n)$, where $n \geq 2$. The empty tuple `()` is the unique value of type `Unit`.

8.1.8 Extractor Patterns

Syntax:

```
Simple_Pattern ::= Stable_Id '(' [Patterns] ')'
```

An extractor pattern $x(p_1, \ldots, p_n)$, where $n \geq 0$, is of the same syntactic form as a constructor pattern. However, instead of a case class, the stable identifier x denotes an object which has a member method named `unapply` or `unapply_sequence` that matches the pattern.

An `unapply` method in an object x *matches* the pattern $x(p_1, \ldots, p_n)$ if it takes exactly one argument and one of the following applies:

> $n = 0$ and `unapply`'s result type is `Boolean`. In this case the extractor pattern matches all values v for which x`.unapply(`v`)` returns **`yes`**.
>
> $n = 1$ and `unapply`'s result type is `Option[`T`]`, for some type T. In this case, the only argument pattern p_1 is typed in turn with expected type T. The extractor pattern matches then all values v for which x`.unapply(`v`)` returns a value of form `Some(`v_1`)`, and p_1 matches v_1.

$n > 1$ and `unapply`'s result type is `Option[(`$T_1, \ldots, T_n$`)]`, for some types $T_1, \ldots, T_n$. In this case, the argument patterns $p_1, \ldots, p_n$ are typed in turn with expected types $T_1, \ldots, T_n$. The extractor pattern matches then all values v for which x`.unapply(`v`)` returns a value of form `Some((`$v_1, \ldots, v_n$`))`, and each pattern p_i matches the corresponding value v_i.

An `unapply_sequence` method in an object x matches the pattern $x(p_1, \ldots, p_n)$, if it takes exactly one argument and its result type is of the form `Option[`S`]`, where S is a subtype of `Sequence[`T`]` for some element type T. This case is further discussed in (§8.1.9).

8.1.9 Pattern Sequences

Syntax:

```
Simple_Pattern ::= Stable_Id '(' [Patterns ',']
                   '*' (variable_id | '_')
                   [',' Patterns] ')'
```

A pattern sequence $p_1, \ldots, p_n$ appears in two contexts. First, in a constructor pattern $c(q_1, \ldots, q_a, p_1, \ldots, p_n, r_1, \ldots, r_b)$, where c is a case class, which has $a + 1 + b$ primary constructor parameters, with a repeated parameter (§4.7.6) of type $*S$ in the middle. Second, in an extractor pattern $x(p_1, \ldots, p_n)$, if the extractor object x has an `unapply_sequence` method with a result type conforming to `Sequence[`T`]`, but does not have an `unapply` method that matches $p_1, \ldots, p_n$. The expected type for the pattern sequence is in each case the type S.

8.1.10 Conjunction Patterns

Syntax:

```
Simple_Pattern ::= Pattern '&' Pattern
```

A conjunction pattern ("and" *pattern*) matches only if both patterns match. Moreover, only the first pattern in a sequence of conjunction patterns may bind variable names, but variables from the other patterns have their scope extended to the following patterns. This behavior is unlike in pattern alternatives, which aren't allowed to bind variable names at all, due to the fact that each alternative may match a completely different structure, whereas a conjunction pattern matches on the same structure.

8.1.11 List Patterns

Syntax:

```
Simple_Pattern ::= Pattern '~>' Pattern
                 | Pattern '<~' Pattern
```

A list pattern is a pattern specialized to match elements of list instances. A list pattern of the form p_1 `~>` p_2 matches a list, where p_1 matches the head element of the list and p_2 matches the remainder of the list, ending with the tail element. A list pattern of the form p_1 `<~` p_2 matches a list, where p_1 matches the tail element of the list and p_2 matches the remained of the list, starting with the head element. If the list has one element, then the matched remainder is **`nil`**.

8.1.12 Range Patterns

Syntax:

```
Simple_Pattern ::= Pattern ('..' | '...') Pattern
```

A range pattern is a pattern specialized to match scalar values with ordering.

8.1.13 Pattern Alternatives

Syntax:

```
Pattern ::= Pattern {'|' Pattern}
```

A pattern alternative p_1 `|` ... `|` p_n, where $n \geq 2$, consists of a number of alternative patterns p_i. All alternative patterns are type checked with the expected type of the pattern. They may not bind variable names other than wildcards (which are discarded). The alternative pattern matches a value v if at least one of its alternatives matches v. Consequently, if a first such match is successful, the remaining patterns are not tested.

Non-normatively: Usually, each pattern alternative is a literal pattern, and therefore binding a variable name makes no sense, since the value that the pattern matching expression matches against is the already bound variable. Still, if needed, the pattern alternatives may be used together with a pattern binder (§8.1.3), which can be useful when the structure that is being matched contains union types.

8.1.14 Regular Expression Patterns

Syntax:

```
Simple_Pattern ::= regexp_literal
```

A regular expression pattern p (*regexp pattern*) is a variant of literal pattern, designed to match `String_Like` values. Literally, the pattern p matches a value v, if v is of

a type that conforms to `String_Like` and its contents match the regular expression. Moreover, sub-patterns bind to variable names of a name of the form `match_`n, where n is either the position of the sub-pattern (unless the sub-pattern is explicitly not captured), or the name of a named sub-pattern.

Regular expression patterns are not designed to match against algebraic data structures.

8.2 Type Patterns

Syntax:

```
Type_Pat ::= Type
```

Type patterns consist of types, type variables and wildcards. A type pattern T is of one of the following forms:

A reference to a class C, p`.`C, p`.type` or T`#`C. This type pattern matches any non-**`nil`** instance of the given class (therefore, it does match the empty tuple `()` with type `Unit`). Note that the prefix of the class, if it is given, is irrelevant for determining class instances, unlike in Scala.

The bottom type `Nothing` (with singleton instance **`nil`**) is the only type pattern that matches **`nil`** (only), but its preferable to match against `Option[`T`]` with implicit conversion of **`nil`** to object `None`.

A singleton type p`.singleton-type`. This type pattern matches only the value denoted by the path p (only the single value denoted by the path p, since **`nil`** is not matched).

A parameterized type pattern T`[`$a_1 \ldots a_n$`]@[`$b_1 \ldots b_n$`][<`$c_1 \ldots c_n$`>]`, where the a_i are type variable patterns or wildcards "`_`", b_i are constraining expressions (§5.8.2) and c_i are unit of measure kinds. This type pattern matches all values which match T for some arbitrary instantiation of the type variables and wildcards.

A compound type pattern T_1 **`with`** $\ldots$ **`with`** T_n, where each T_i is a type pattern. This type pattern matches all values that are matched by each of the type patterns T_i, and in this sense it is equivalent to the pattern T_1 `&` $\ldots$ `&` T_n.

Types are not subject to any type erasure, so it is basically safe to use any other type as type pattern, unlike in Scala.

A *type variable pattern* is a simple identifier which starts with a lower case letter.

8.3 Pattern Matching Expressions

Syntax:

```
Match_Expr     ::= Pat_Match_Expr | Case_Expr
Pat_Match_Expr ::= 'match' Simple_Expr1 Match_Body
Match_Body     ::= semi When_Clauses 'end' ['match']
                 | '{' When_Clauses '}'
When_Clauses   ::= When_Clause {['next'] semi When_Clause}
                   [['next'] 'else' Block]
When_Clause    ::= 'when' Pattern [Guard] ('then' | semi) Block
```

A pattern matching expression

match e { **when** p_1 **then** b_1 ... **when** p_n **then** b_n **else** b_{n+1} }

consists of a selector expression e, a number $n > 0$ of cases and an optional "default" case, denoted with **else** keyword. Each case consists of a (possibly guarded) pattern p_i and a block b_i, or just the block b_{n+1}. Each p_i might be complemented by a guard **if** e or **unless** e, where e is a the guarding expression that is evaluated as boolean. The scope of the pattern variables in p_i comprises the pattern's guard and the corresponding block b_i.

8.4 Pattern Matching Anonymous Functions

Syntax:

```
Block_Expr ::= '{' When_Clauses '}'
```

Chapter 9

Top-Level Definitions

Chapter 10

Annotations

Chapter 11

Naming Guidelines

Chapter 12

The Coral Standard Library

12.1 Root Classes

12.1.1 The Object Class

12.1.2 The Nothing Class

12.2 Value Classes

12.3 Standard Reference Classes

Chapter A

Coral Syntax Summary